Rules for integrands involving $(a + b \text{ArcTan}[c\,x])^p$

4. $\int (f\,x)^m (d + e\,x)^q (a + b \text{ArcTan}[c\,x])^p\, dx$ when $p \in \mathbb{Z}^+$

1. $\int \frac{(f\,x)^m (a + b \text{ArcTan}[c\,x])^p}{d + e\,x}\, dx$ when $p \in \mathbb{Z}^+ \wedge c^2 d^2 + e^2 == 0$

1: $\int \frac{(f\,x)^m (a + b \text{ArcTan}[c\,x])^p}{d + e\,x}\, dx$ when $p \in \mathbb{Z}^+ \wedge c^2 d^2 + e^2 == 0 \wedge m > 0$

- **Derivation: Algebraic expansion**
- **Basis:** $\frac{x}{d+e\,x} == \frac{1}{e} - \frac{d}{e\,(d+e\,x)}$
- **Rule: If** $p \in \mathbb{Z}^+ \wedge c^2 d^2 + e^2 == 0 \wedge m > 0$**, then**

$$\int \frac{(f\,x)^m (a + b \text{ArcTan}[c\,x])^p}{d + e\,x}\, dx \longrightarrow \frac{f}{e} \int (f\,x)^{m-1} (a + b \text{ArcTan}[c\,x])^p\, dx - \frac{d\,f}{e} \int \frac{(f\,x)^{m-1} (a + b \text{ArcTan}[c\,x])^p}{d + e\,x}\, dx$$

- **Program code:**

```
Int[(f_.*x_)^m_.*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_),x_Symbol] :=
  f/e*Int[(f*x)^(m-1)*(a+b*ArcTan[c*x])^p,x] -
  d*f/e*Int[(f*x)^(m-1)*(a+b*ArcTan[c*x])^p/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && EqQ[c^2*d^2+e^2,0] && GtQ[m,0]
```

```
Int[(f_.*x_)^m_.*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_),x_Symbol] :=
  f/e*Int[(f*x)^(m-1)*(a+b*ArcCot[c*x])^p,x] -
  d*f/e*Int[(f*x)^(m-1)*(a+b*ArcCot[c*x])^p/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && EqQ[c^2*d^2+e^2,0] && GtQ[m,0]
```

2. $\int \frac{(f\,x)^m (a + b \text{ArcTan}[c\,x])^p}{d + e\,x}\, dx$ when $p \in \mathbb{Z}^+ \wedge c^2 d^2 + e^2 == 0 \wedge m < 0$

1: $\int \frac{(a + b \text{ArcTan}[c\,x])^p}{x\,(d + e\,x)}\, dx$ when $p \in \mathbb{Z}^+ \wedge c^2 d^2 + e^2 == 0$

- **Derivation: Integration by parts**
- **Basis:** $\frac{1}{x\,(d+e\,x)} == \frac{1}{d}\,\partial_x \text{Log}\left[2 - \frac{2}{1+\frac{e\,x}{d}}\right]$
- **Rule: If** $p \in \mathbb{Z}^+ \wedge c^2 d^2 + e^2 == 0$**, then**

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\,(d+e\,x)}\,dx \longrightarrow \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{Log}\left[2-\frac{2}{1+\frac{e\,x}{d}}\right]}{d}-\frac{b\,c\,p}{d}\int \frac{(a+b\,\text{ArcTan}[c\,x])^{p-1}\,\text{Log}\left[2-\frac{2}{1+\frac{e\,x}{d}}\right]}{1+c^2\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./(x_*(d_+e_.*x_)),x_Symbol] :=
  (a+b*ArcTan[c*x])^p*Log[2-2/(1+e*x/d)]/d -
  b*c*p/d*Int[(a+b*ArcTan[c*x])^(p-1)*Log[2-2/(1+e*x/d)]/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[c^2*d^2+e^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./(x_*(d_+e_.*x_)),x_Symbol] :=
  (a+b*ArcCot[c*x])^p*Log[2-2/(1+e*x/d)]/d +
  b*c*p/d*Int[(a+b*ArcCot[c*x])^(p-1)*Log[2-2/(1+e*x/d)]/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[c^2*d^2+e^2,0]
```

2: $\int \frac{(f\,x)^m\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x}\,dx$ when $p\in\mathbb{Z}^+\land c^2\,d^2+e^2==0\land m<-1$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e\,x}==\frac{1}{d}-\frac{e\,x}{d\,(d+e\,x)}$

Rule: If $p\in\mathbb{Z}^+\land c^2\,d^2+e^2==0\land m<-1$**, then**

$$\int \frac{(f\,x)^m\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x}\,dx \longrightarrow \frac{1}{d}\int (f\,x)^m\,(a+b\,\text{ArcTan}[c\,x])^p\,dx-\frac{e}{d\,f}\int \frac{(f\,x)^{m+1}\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_),x_Symbol] :=
  1/d*Int[(f*x)^m*(a+b*ArcTan[c*x])^p,x] -
  e/(d*f)*Int[(f*x)^(m+1)*(a+b*ArcTan[c*x])^p/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && EqQ[c^2*d^2+e^2,0] && LtQ[m,-1]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_),x_Symbol] :=
  1/d*Int[(f*x)^m*(a+b*ArcCot[c*x])^p,x] -
  e/(d*f)*Int[(f*x)^(m+1)*(a+b*ArcCot[c*x])^p/(d+e*x),x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && EqQ[c^2*d^2+e^2,0] && LtQ[m,-1]
```

2: $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{ArcTan}[c\,x])\,dx$ when $q \neq -1 \land 2\,m \in \mathbb{Z} \land ((m \mid q) \in \mathbb{Z}^+ \lor m+q+1 \in \mathbb{Z}^- \land m\,q < 0)$

Derivation: Integration by parts

Rule: If $q \neq -1 \land 2\,m \in \mathbb{Z} \land ((m \mid q) \in \mathbb{Z}^+ \lor m+q+1 \in \mathbb{Z}^- \land m\,q < 0)$**, let** $u \to \int (f\,x)^m\,(d+e\,x)^q\,dx$**, then**

$$\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{ArcTan}[c\,x])\,dx \longrightarrow u\,(a+b\,\text{ArcTan}[c\,x]) - b\,c\int \frac{u}{1+c^2\,x^2}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_)^q_.*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x)^q,x]},
  Dist[(a+b*ArcTan[c*x]),u] - b*c*Int[SimplifyIntegrand[u/(1+c^2*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,q},x] && NeQ[q,-1] && IntegerQ[2*m] && (IGtQ[m,0] && IGtQ[q,0] || ILtQ[m+q+1,0] && LtQ[m*q,0])
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_)^q_.*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x)^q,x]},
  Dist[(a+b*ArcCot[c*x]),u] + b*c*Int[SimplifyIntegrand[u/(1+c^2*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,q},x] && NeQ[q,-1] && IntegerQ[2*m] && (IGtQ[m,0] && IGtQ[q,0] || ILtQ[m+q+1,0] && LtQ[m*q,0])
```

3: $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $p-1 \in \mathbb{Z}^+ \land c^2\,d^2+e^2 == 0 \land (m \mid q) \in \mathbb{Z} \land q \neq -1$

Derivation: Integration by parts

Rule: If $p-1 \in \mathbb{Z}^+ \land c^2\,d^2+e^2 == 0 \land (m \mid q) \in \mathbb{Z} \land q \neq -1$**, let** $u \to \int (f\,x)^m\,(d+e\,x)^q\,dx$**, then**

$$\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{ArcTan}[c\,x])^p\,dx \longrightarrow$$

$$u\,(a+b\,\text{ArcTan}[c\,x])^p - b\,c\,p\int (a+b\,\text{ArcTan}[c\,x])^{p-1}\,\text{ExpandIntegrand}\left[\frac{u}{1+c^2\,x^2},\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_,x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x)^q,x]},
  Dist[(a+b*ArcTan[c*x])^p,u] - b*c*p*Int[ExpandIntegrand[(a+b*ArcTan[c*x])^(p-1),u/(1+c^2*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,q},x] && IGtQ[p,1] && EqQ[c^2*d^2+e^2,0] && IntegersQ[m,q] && NeQ[m,-1] && NeQ[q,-1] && ILtQ[m+q+1,0] && LtQ[
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_,x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x)^q,x]},
  Dist[(a+b*ArcCot[c*x])^p,u] + b*c*p*Int[ExpandIntegrand[(a+b*ArcCot[c*x])^(p-1),u/(1+c^2*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,q},x] && IGtQ[p,1] && EqQ[c^2*d^2+e^2,0] && IntegersQ[m,q] && NeQ[m,-1] && NeQ[q,-1] && ILtQ[m+q+1,0] && LtQ[
```

4: $\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $p\in\mathbb{Z}^+\land q\in\mathbb{Z}\land(q>0\lor a\neq 0\lor m\in\mathbb{Z})$

Rule: If $p\in\mathbb{Z}^+\land q\in\mathbb{Z}\land(q>0\lor a\neq 0\lor m\in\mathbb{Z})$**, then**

$$\int (f\,x)^m\,(d+e\,x)^q\,(a+b\,\text{ArcTan}[c\,x])^p\,dx\,\longrightarrow\,\int (a+b\,\text{ArcTan}[c\,x])^p\,\text{ExpandIntegrand}[(f\,x)^m\,(d+e\,x)^q,\,x]\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcTan[c*x])^p,(f*x)^m*(d+e*x)^q,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0] && IntegerQ[q] && (GtQ[q,0] || NeQ[a,0] || IntegerQ[m])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCot[c*x])^p,(f*x)^m*(d+e*x)^q,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0] && IntegerQ[q] && (GtQ[q,0] || NeQ[a,0] || IntegerQ[m])
```

5. $\int (d + e x^2)^q (a + b \operatorname{ArcTan}[c x])^p \, dx$

1. $\int (d + e x^2)^q (a + b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d$

1. $\int (d + e x^2)^q (a + b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \wedge q > 0$

1: $\int (d + e x^2)^q (a + b \operatorname{ArcTan}[c x]) \, dx$ when $e == c^2 d \wedge q > 0$

Rule: If $e == c^2 d \wedge q > 0$**, then**

$$\int (d + e x^2)^q (a + b \operatorname{ArcTan}[c x]) \, dx \longrightarrow -\frac{b (d + e x^2)^q}{2 c q (2 q + 1)} + \frac{x (d + e x^2)^q (a + b \operatorname{ArcTan}[c x])}{2 q + 1} + \frac{2 d q}{2 q + 1} \int (d + e x^2)^{q-1} (a + b \operatorname{ArcTan}[c x]) \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  -b*(d+e*x^2)^q/(2*c*q*(2*q+1)) +
  x*(d+e*x^2)^q*(a+b*ArcTan[c*x])/(2*q+1) +
  2*d*q/(2*q+1)*Int[(d+e*x^2)^(q-1)*(a+b*ArcTan[c*x]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[q,0]
```

```
Int[(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  b*(d+e*x^2)^q/(2*c*q*(2*q+1)) +
  x*(d+e*x^2)^q*(a+b*ArcCot[c*x])/(2*q+1) +
  2*d*q/(2*q+1)*Int[(d+e*x^2)^(q-1)*(a+b*ArcCot[c*x]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[q,0]
```

2: $\int \left(d+e x^2\right)^q (a+b \text{ArcTan}[c x])^p \, dx$ when $e == c^2 d \wedge q > 0 \wedge p > 1$

Rule: If $e == c^2 d \wedge q > 0 \wedge p > 1$**, then**

$$\int \left(d+e x^2\right)^q (a+b \text{ArcTan}[c x])^p \, dx \longrightarrow$$

$$-\frac{b p \left(d+e x^2\right)^q (a+b \text{ArcTan}[c x])^{p-1}}{2 c q (2 q+1)}+\frac{x \left(d+e x^2\right)^q (a+b \text{ArcTan}[c x])^p}{2 q+1}+$$
$$\frac{2 d q}{2 q+1} \int \left(d+e x^2\right)^{q-1} (a+b \text{ArcTan}[c x])^p \, dx+\frac{b^2 d p (p-1)}{2 q (2 q+1)} \int \left(d+e x^2\right)^{q-1} (a+b \text{ArcTan}[c x])^{p-2} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_,x_Symbol] :=
  -b*p*(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p-1)/(2*c*q*(2*q+1)) +
  x*(d+e*x^2)^q*(a+b*ArcTan[c*x])^p/(2*q+1) +
  2*d*q/(2*q+1)*Int[(d+e*x^2)^(q-1)*(a+b*ArcTan[c*x])^p,x] +
  b^2*d*p*(p-1)/(2*q*(2*q+1))*Int[(d+e*x^2)^(q-1)*(a+b*ArcTan[c*x])^(p-2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[q,0] && GtQ[p,1]
```

```
Int[(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_,x_Symbol] :=
  b*p*(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p-1)/(2*c*q*(2*q+1)) +
  x*(d+e*x^2)^q*(a+b*ArcCot[c*x])^p/(2*q+1) +
  2*d*q/(2*q+1)*Int[(d+e*x^2)^(q-1)*(a+b*ArcCot[c*x])^p,x] +
  b^2*d*p*(p-1)/(2*q*(2*q+1))*Int[(d+e*x^2)^(q-1)*(a+b*ArcCot[c*x])^(p-2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[q,0] && GtQ[p,1]
```

2. $\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx$ when $e == c^2 d \wedge q < 0$

1. $\int \frac{(a+b \text{ArcTan}[c x])^p}{d+e x^2} \, dx$ when $e == c^2 d$

x: $\int \frac{(a+b \text{ArcTan}[c x])^p}{d+e x^2} \, dx$ when $e == c^2 d$

Derivation: Integration by substitution

Basis: If $e == c^2 d$**, then** $\frac{F[\text{ArcTan}[c x]]}{d+e x^2} == \frac{1}{c d} \text{Subst}[F[x], x, \text{ArcTan}[c x]] \, \partial_x \text{ArcTan}[c x]$

Rule: If $e == c^2 d$**, then**

$$\int \frac{(a+b \text{ArcTan}[c x])^p}{d+e x^2} \, dx \rightarrow \frac{1}{c d} \text{Subst}\left[\int (a+b x)^p \, dx, x, \text{ArcTan}[c x]\right]$$

Program code:

```
(* Int[(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  1/(c*d)*Subst[Int[(a+b*x)^p,x],x,ArcTan[c*x]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] *)
```

```
(* Int[(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  -1/(c*d)*Subst[Int[(a+b*x)^p,x],x,ArcCot[c*x]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] *)
```

1: $\int \frac{1}{(d+e x^2)(a+b \text{ArcTan}[c x])} \, dx$ when $e == c^2 d$

Derivation: Integration by substitution

Rule: If $e == c^2 d$**, then**

$$\int \frac{1}{(d+e x^2)(a+b \text{ArcTan}[c x])} \, dx \rightarrow \frac{\text{Log}[a+b \text{ArcTan}[c x]]}{b c d}$$

Program code:

```
Int[1/((d_+e_.*x_^2)*(a_.+b_.*ArcTan[c_.*x_])),x_Symbol] :=
  Log[RemoveContent[a+b*ArcTan[c*x],x]]/(b*c*d) /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d]
```

```
Int[1/((d_+e_.*x_^2)*(a_.+b_.*ArcCot[c_.*x_])),x_Symbol] :=
  -Log[RemoveContent[a+b*ArcCot[c*x],x]]/(b*c*d) /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d]
```

2: $\int \frac{(a + b \operatorname{ArcTan}[c x])^p}{d + e x^2} \, dx$ when $e == c^2 d \wedge p \neq -1$

Derivation: Integration by substitution

– **Rule: If** $e == c^2 d \wedge p \neq -1$**, then**

$$\int \frac{(a + b \operatorname{ArcTan}[c x])^p}{d + e x^2} \, dx \longrightarrow \frac{(a + b \operatorname{ArcTan}[c x])^{p+1}}{b c d (p + 1)}$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  (a+b*ArcTan[c*x])^(p+1)/(b*c*d*(p+1)) /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && NeQ[p,-1]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  -(a+b*ArcCot[c*x])^(p+1)/(b*c*d*(p+1)) /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && NeQ[p,-1]
```

2. $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2 d \wedge p \in \mathbb{Z}^+$

1. $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2 d \wedge n \in \mathbb{Z}^+ \wedge d > 0$

1: $\int \frac{(a+b\,\text{ArcTan}[c\,x])}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2 d \wedge d > 0$

Derivation: Integration by substitution and algebraic simplification

Note: Although not essential, these rules returns antiderivatives free of complex exponentials of the form $i\,e^{\text{ArcTan}[c\,x]}$ **and** $e^{\text{ArcCot}[c\,x]}$**.**

Basis: If $e == c^2 d \wedge d > 0$**, then** $\frac{1}{\sqrt{d+e\,x^2}} == \frac{1}{c\sqrt{d}}\,\text{Sec}[\text{ArcTan}[c\,x]]\,\partial_x \text{ArcTan}[c\,x]$

Basis: If $e == c^2 d \wedge d > 0$**, then** $\frac{1}{\sqrt{d+e\,x^2}} == -\frac{1}{c\sqrt{d}}\sqrt{\text{Csc}[\text{ArcCot}[c\,x]]^2}\,\partial_x \text{ArcCot}[c\,x]$

Rule: If $e == c^2 d \wedge d > 0$**, then**

$$\int \frac{a+b\,\text{ArcTan}[c\,x]}{\sqrt{d+e\,x^2}}\,dx \to \frac{1}{c\sqrt{d}}\,\text{Subst}[(a+b\,x)\,\text{Sec}[x],\,x,\,\text{ArcTan}[c\,x]]$$

$$\to -\frac{2\,i\,(a+b\,\text{ArcTan}[c\,x])\,\text{ArcTan}\left[\frac{\sqrt{1+i\,c\,x}}{\sqrt{1-i\,c\,x}}\right]}{c\sqrt{d}} + \frac{i\,b\,\text{PolyLog}\left[2,\,-\frac{i\sqrt{1+i\,c\,x}}{\sqrt{1-i\,c\,x}}\right]}{c\sqrt{d}} - \frac{i\,b\,\text{PolyLog}\left[2,\,\frac{i\sqrt{1+i\,c\,x}}{\sqrt{1-i\,c\,x}}\right]}{c\sqrt{d}}$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  -2*I*(a+b*ArcTan[c*x])*ArcTan[Sqrt[1+I*c*x]/Sqrt[1-I*c*x]]/(c*Sqrt[d]) +
  I*b*PolyLog[2,-I*Sqrt[1+I*c*x]/Sqrt[1-I*c*x]]/(c*Sqrt[d]) -
  I*b*PolyLog[2,I*Sqrt[1+I*c*x]/Sqrt[1-I*c*x]]/(c*Sqrt[d]) /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[d,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  -2*I*(a+b*ArcCot[c*x])*ArcTan[Sqrt[1+I*c*x]/Sqrt[1-I*c*x]]/(c*Sqrt[d]) -
  I*b*PolyLog[2,-I*Sqrt[1+I*c*x]/Sqrt[1-I*c*x]]/(c*Sqrt[d]) +
  I*b*PolyLog[2,I*Sqrt[1+I*c*x]/Sqrt[1-I*c*x]]/(c*Sqrt[d]) /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[d,0]
```

2. $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2 d \land p \in \mathbb{Z}^+ \land d > 0$

1: $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2 d \land p \in \mathbb{Z}^+ \land d > 0$

Derivation: Integration by substitution

- **Basis: If** $e == c^2 d \land d > 0$**, then** $\frac{1}{\sqrt{d+e\,x^2}} == \frac{1}{c\,\sqrt{d}}\,\text{Sec}[\text{ArcTan}[c\,x]]\,\partial_x \text{ArcTan}[c\,x]$

- **Rule: If** $e == c^2 d \land p \in \mathbb{Z}^+ \land d > 0$**, then**

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx \longrightarrow \frac{1}{c\,\sqrt{d}}\,\text{Subst}\Big[\int (a+b\,x)^p\,\text{Sec}[x]\,dx,\ x,\ \text{ArcTan}[c\,x]\Big]$$

- **Program code:**

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  1/(c*Sqrt[d])*Subst[Int[(a+b*x)^p*Sec[x],x],x,ArcTan[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && GtQ[d,0]
```

2: $\int \frac{(a+b\,\text{ArcCot}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2 d \land p \in \mathbb{Z}^+ \land d > 0$

Derivation: Integration by substitution and piecewise constant extraction

- **Basis: If** $e == c^2 d \land d > 0$**, then** $\frac{1}{\sqrt{d+e\,x^2}} == -\frac{1}{c\,\sqrt{d}}\,\frac{\text{Csc}[\text{ArcCot}[c\,x]]^2}{\sqrt{\text{Csc}[\text{ArcCot}[c\,x]]^2}}\,\partial_x \text{ArcCot}[c\,x]$

- **Basis:** $\partial_x \frac{\text{Csc}[x]}{\sqrt{\text{Csc}[x]^2}} == 0$

- **Basis:** $\frac{\text{Csc}[\text{ArcCot}[c\,x]]}{\sqrt{\text{Csc}[\text{ArcCot}[c\,x]]^2}} == \frac{c\,x\,\sqrt{1+\frac{1}{c^2 x^2}}}{\sqrt{1+c^2 x^2}}$

- **Rule: If** $e == c^2 d \land p \in \mathbb{Z}^+ \land d > 0$**, then**

$$\int \frac{(a+b\,\text{ArcCot}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx \longrightarrow -\frac{1}{c\,\sqrt{d}}\,\text{Subst}\Big[\int \frac{(a+b\,x)^p\,\text{Csc}[x]^2}{\sqrt{\text{Csc}[x]^2}}\,dx,\ x,\ \text{ArcCot}[c\,x]\Big]$$

$$\to -\frac{x\sqrt{1+\frac{1}{c^2 x^2}}}{\sqrt{d+e x^2}} \text{Subst}\left[\int (a+b x)^p \text{Csc}[x] \, dx, x, \text{ArcCot}[c x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  -x*Sqrt[1+1/(c^2*x^2)]/Sqrt[d+e*x^2]*Subst[Int[(a+b*x)^p*Csc[x],x],x,ArcCot[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && GtQ[d,0]
```

2: $\int \frac{(a+b \text{ArcTan}[c x])^p}{\sqrt{d+e x^2}} \, dx$ when $e = c^2 d \wedge p \in \mathbb{Z}^+ \wedge d \ngtr 0$

Derivation: Piecewise constant extraction

- **Basis: If** $e = c^2 d$, **then** $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} = 0$

Rule: If $e = c^2 d \wedge p \in \mathbb{Z}^+ \wedge d \ngtr 0$, **then**

$$\int \frac{(a+b \text{ArcTan}[c x])^p}{\sqrt{d+e x^2}} \, dx \to \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} \int \frac{(a+b \text{ArcTan}[c x])^p}{\sqrt{1+c^2 x^2}} \, dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[(a+b*ArcTan[c*x])^p/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && Not[GtQ[d,0]]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[(a+b*ArcCot[c*x])^p/Sqrt[1+c^2*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && Not[GtQ[d,0]]
```

3. $\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx$ when $e = c^2 d \wedge q < -1$

1: $\int \frac{(a+b \text{ArcTan}[c x])^p}{(d+e x^2)^2} \, dx$ when $e = c^2 d \wedge p > 0$

Rule: If $e = c^2 d \wedge p > 0$, **then**

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{\left(d+e\,x^2\right)^2}\,dx \longrightarrow \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{2\,d\,\left(d+e\,x^2\right)}+\frac{(a+b\,\text{ArcTan}[c\,x])^{p+1}}{2\,b\,c\,d^2\,(p+1)}-\frac{b\,c\,p}{2}\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^{p-1}}{\left(d+e\,x^2\right)^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2)^2,x_Symbol] :=
  x*(a+b*ArcTan[c*x])^p/(2*d*(d+e*x^2)) +
  (a+b*ArcTan[c*x])^(p+1)/(2*b*c*d^2*(p+1)) -
  b*c*p/2*Int[x*(a+b*ArcTan[c*x])^(p-1)/(d+e*x^2)^2,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2)^2,x_Symbol] :=
  x*(a+b*ArcCot[c*x])^p/(2*d*(d+e*x^2)) -
  (a+b*ArcCot[c*x])^(p+1)/(2*b*c*d^2*(p+1)) +
  b*c*p/2*Int[x*(a+b*ArcCot[c*x])^(p-1)/(d+e*x^2)^2,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,0]
```

2. $\int \left(d+e\,x^2\right)^q\,(a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $e == c^2\,d \land q < -1 \land p \geq 1$

   1. $\int \left(d+e\,x^2\right)^q\,(a+b\,\text{ArcTan}[c\,x])\,dx$ when $e == c^2\,d \land q < -1$

      1: $\int \frac{a+b\,\text{ArcTan}[c\,x]}{\left(d+e\,x^2\right)^{3/2}}\,dx$ when $e == c^2\,d$

Rule: If $e == c^2\,d$, then

$$\int \frac{a+b\,\text{ArcTan}[c\,x]}{\left(d+e\,x^2\right)^{3/2}}\,dx \longrightarrow \frac{b}{c\,d\,\sqrt{d+e\,x^2}}+\frac{x\,(a+b\,\text{ArcTan}[c\,x])}{d\,\sqrt{d+e\,x^2}}$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])/(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  b/(c*d*Sqrt[d+e*x^2]) +
  x*(a+b*ArcTan[c*x])/(d*Sqrt[d+e*x^2]) /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])/(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  -b/(c*d*Sqrt[d+e*x^2]) +
  x*(a+b*ArcCot[c*x])/(d*Sqrt[d+e*x^2]) /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d]
```

2: $\int (d+e x^2)^q (a+b \text{ArcTan}[c x]) \, dx$ when $e == c^2 d \bigwedge q < -1 \bigwedge q \neq -\frac{3}{2}$

Rule: If $e == c^2 d \bigwedge q < -1 \bigwedge q \neq -\frac{3}{2}$, **then**

$$\int (d+e x^2)^q (a+b \text{ArcTan}[c x]) \, dx \longrightarrow \frac{b (d+e x^2)^{q+1}}{4 c d (q+1)^2} - \frac{x (d+e x^2)^{q+1} (a+b \text{ArcTan}[c x])}{2 d (q+1)} + \frac{2 q+3}{2 d (q+1)} \int (d+e x^2)^{q+1} (a+b \text{ArcTan}[c x]) \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  b*(d+e*x^2)^(q+1)/(4*c*d*(q+1)^2) -
  x*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])/(2*d*(q+1)) +
  (2*q+3)/(2*d*(q+1))*Int[(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[q,-1] && NeQ[q,-3/2]
```

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  -b*(d+e*x^2)^(q+1)/(4*c*d*(q+1)^2) -
  x*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])/(2*d*(q+1)) +
  (2*q+3)/(2*d*(q+1))*Int[(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[q,-1] && NeQ[q,-3/2]
```

2. $\int (d + e x^2)^q (a + b \text{ArcTan}[c x])^p \, dx$ when $e == c^2 d \wedge q < -1 \wedge p > 1$

1: $\int \frac{(a + b \text{ArcTan}[c x])^p}{(d + e x^2)^{3/2}} \, dx$ when $e == c^2 d \wedge p > 1$

Rule: If $e == c^2 d \wedge p > 1$, then

$$\int \frac{(a + b \text{ArcTan}[c x])^p}{(d + e x^2)^{3/2}} \, dx \longrightarrow \frac{b p (a + b \text{ArcTan}[c x])^{p-1}}{c d \sqrt{d + e x^2}} + \frac{x (a + b \text{ArcTan}[c x])^p}{d \sqrt{d + e x^2}} - b^2 p (p - 1) \int \frac{(a + b \text{ArcTan}[c x])^{p-2}}{(d + e x^2)^{3/2}} \, dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_/(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  b*p*(a+b*ArcTan[c*x])^(p-1)/(c*d*Sqrt[d+e*x^2]) +
  x*(a+b*ArcTan[c*x])^p/(d*Sqrt[d+e*x^2]) -
  b^2*p*(p-1)*Int[(a+b*ArcTan[c*x])^(p-2)/(d+e*x^2)^(3/2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,1]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_/(d_+e_.*x_^2)^(3/2),x_Symbol] :=
  -b*p*(a+b*ArcCot[c*x])^(p-1)/(c*d*Sqrt[d+e*x^2]) +
  x*(a+b*ArcCot[c*x])^p/(d*Sqrt[d+e*x^2]) -
  b^2*p*(p-1)*Int[(a+b*ArcCot[c*x])^(p-2)/(d+e*x^2)^(3/2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,1]
```

2: $\int (d + e x^2)^q (a + b \text{ArcTan}[c x])^p \, dx$ when $e == c^2 d \wedge q < -1 \wedge p > 1 \wedge q \neq -\frac{3}{2}$

- **Rule: If** $e == c^2 d \wedge q < -1 \wedge p > 1 \wedge q \neq -\frac{3}{2}$**, then**

$$\int (d + e x^2)^q (a + b \text{ArcTan}[c x])^p \, dx \longrightarrow$$

$$\frac{b p (d + e x^2)^{q+1} (a + b \text{ArcTan}[c x])^{p-1}}{4 c d (q+1)^2} - \frac{x (d + e x^2)^{q+1} (a + b \text{ArcTan}[c x])^p}{2 d (q+1)} +$$

$$\frac{2 q + 3}{2 d (q+1)} \int (d + e x^2)^{q+1} (a + b \text{ArcTan}[c x])^p \, dx - \frac{b^2 p (p-1)}{4 (q+1)^2} \int (d + e x^2)^q (a + b \text{ArcTan}[c x])^{p-2} \, dx$$

- **Program code:**

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_,x_Symbol] :=
  b*p*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^(p-1)/(4*c*d*(q+1)^2) -
  x*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^p/(2*d*(q+1)) +
  (2*q+3)/(2*d*(q+1))*Int[(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^p,x] -
  b^2*p*(p-1)/(4*(q+1)^2)*Int[(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p-2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[q,-1] && GtQ[p,1] && NeQ[q,-3/2]
```

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_,x_Symbol] :=
  -b*p*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^(p-1)/(4*c*d*(q+1)^2) -
  x*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^p/(2*d*(q+1)) +
  (2*q+3)/(2*d*(q+1))*Int[(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^p,x] -
  b^2*p*(p-1)/(4*(q+1)^2)*Int[(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p-2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[q,-1] && GtQ[p,1] && NeQ[q,-3/2]
```

3: $\int (d + e x^2)^q (a + b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \land q < -1 \land p < -1$

Derivation: Integration by parts

- Basis: If $e == c^2 d$, then $\frac{(a+b \operatorname{ArcTan}[c x])^p}{d+e x^2} == \partial_x \frac{(a+b \operatorname{ArcTan}[c x])^{p+1}}{b c d (p+1)}$

- Rule: If $e == c^2 d \land q < -1 \land p < -1$, then

$$\int (d + e x^2)^q (a + b \operatorname{ArcTan}[c x])^p \, dx \longrightarrow \frac{(d + e x^2)^{q+1} (a + b \operatorname{ArcTan}[c x])^{p+1}}{b c d (p + 1)} - \frac{2 c (q + 1)}{b (p + 1)} \int x (d + e x^2)^q (a + b \operatorname{ArcTan}[c x])^{p+1} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_,x_Symbol] :=
  (d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^(p+1)/(b*c*d*(p+1)) -
  2*c*(q+1)/(b*(p+1))*Int[x*(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[q,-1] && LtQ[p,-1]
```

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_,x_Symbol] :=
  -(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^(p+1)/(b*c*d*(p+1)) +
  2*c*(q+1)/(b*(p+1))*Int[x*(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[q,-1] && LtQ[p,-1]
```

4. $\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx$ when $e = c^2 d \wedge 2(q+1) \in \mathbb{Z}^-$

1. $\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx$ when $e = c^2 d \wedge 2(q+1) \in \mathbb{Z}^-$

1: $\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx$ when $e = c^2 d \wedge 2(q+1) \in \mathbb{Z}^- \wedge (q \in \mathbb{Z} \vee d > 0)$

Derivation: Integration by substitution

Basis: If $e = c^2 d \wedge 2(q+1) \in \mathbb{Z} \wedge (q \in \mathbb{Z} \vee d > 0)$**, then** $(d+e x^2)^q = \frac{d^q}{c \text{Cos}[\text{ArcTan}[c x]]^{2(q+1)}} \partial_x \text{ArcTan}[c x]$

Rule: If $e = c^2 d \wedge 2(q+1) \in \mathbb{Z}^- \wedge (q \in \mathbb{Z} \vee d > 0)$**, then**

$$\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx \longrightarrow \frac{d^q}{c} \text{Subst}\left[\int \frac{(a+b x)^p}{\text{Cos}[x]^{2(q+1)}} \, dx, x, \text{ArcTan}[c x]\right]$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  d^q/c*Subst[Int[(a+b*x)^p/Cos[x]^(2*(q+1)),x],x,ArcTan[c*x]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && ILtQ[2*(q+1),0] && (IntegerQ[q] || GtQ[d,0])
```

2: $\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx$ when $e = c^2 d \wedge 2(q+1) \in \mathbb{Z}^- \wedge \neg (q \in \mathbb{Z} \vee d > 0)$

Derivation: Piecewise constant extraction

Basis: If $e = c^2 d$**, then** $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} = 0$

Rule: If $e = c^2 d \wedge 2(q+1) \in \mathbb{Z}^- \wedge \neg (q \in \mathbb{Z} \vee d > 0)$**, then**

$$\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx \longrightarrow \frac{d^{q+\frac{1}{2}} \sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} \int (1+c^2 x^2)^q (a+b \text{ArcTan}[c x])^p \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  d^(q+1/2)*Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[(1+c^2*x^2)^q*(a+b*ArcTan[c*x])^p,x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && ILtQ[2*(q+1),0] && Not[IntegerQ[q] || GtQ[d,0]]
```

2. $\int (d+e x^2)^q (a+b \text{ArcCot}[c x])^p \, dx$ when $e = c^2 d \land 2 (q+1) \in \mathbb{Z}^-$

1: $\int (d+e x^2)^q (a+b \text{ArcCot}[c x])^p \, dx$ when $e = c^2 d \land 2 (q+1) \in \mathbb{Z}^- \land q \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $e = c^2 d \land q \in \mathbb{Z}$**, then** $(d+e x^2)^q = -\frac{d^q}{c \, \text{Sin}[\text{ArcCot}[c x]]^{2 (q+1)}} \partial_x \text{ArcCot}[c x]$

Rule: If $e = c^2 d \land 2 (q+1) \in \mathbb{Z}^- \land q \in \mathbb{Z}$**, then**

$$\int (d+e x^2)^q (a+b \text{ArcCot}[c x])^p \, dx \to -\frac{d^q}{c} \text{Subst}\left[\int \frac{(a+b x)^p}{\text{Sin}[x]^{2 (q+1)}} \, dx, x, \text{ArcCot}[c x]\right]$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  -d^q/c*Subst[Int[(a+b*x)^p/Sin[x]^(2*(q+1)),x],x,ArcCot[c*x]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && ILtQ[2*(q+1),0] && IntegerQ[q]
```

2: $\int (d+e x^2)^q (a+b\,\text{ArcCot}[c x])^p\,dx$ when $e = c^2 d \land 2(q+1) \in \mathbb{Z}^- \land q \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: If $e = c^2 d$, then $\partial_x \frac{x\sqrt{\frac{1+c^2 x^2}{c^2 x^2}}}{\sqrt{d+e x^2}} = 0$
- Basis: If $2(q+1) \in \mathbb{Z} \land q \notin \mathbb{Z}$, then $x\sqrt{1+\frac{1}{c^2 x^2}}\,(1+c^2 x^2)^{q-\frac{1}{2}} = -\frac{1}{c^2\,\text{Sin}[\text{ArcCot}[c x]]^{2(q+1)}}\,\partial_x \text{ArcCot}[c x]$
- Rule: If $e = c^2 d \land 2(q+1) \in \mathbb{Z}^- \land q \notin \mathbb{Z}$, then

$$\int (d+e x^2)^q (a+b\,\text{ArcCot}[c x])^p\,dx \to \frac{c^2 d^{q+\frac{1}{2}}\,x\sqrt{\frac{1+c^2 x^2}{c^2 x^2}}}{\sqrt{d+e x^2}} \int x\sqrt{1+\frac{1}{c^2 x^2}}\,(1+c^2 x^2)^{q-\frac{1}{2}} (a+b\,\text{ArcCot}[c x])^p\,dx$$

$$\to -\frac{d^{q+\frac{1}{2}}\,x\sqrt{\frac{1+c^2 x^2}{c^2 x^2}}}{\sqrt{d+e x^2}}\,\text{Subst}\Big[\int \frac{(a+b x)^p}{\text{Sin}[x]^{2(q+1)}}\,dx,\ x,\ \text{ArcCot}[c x]\Big]$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  -d^(q+1/2)*x*Sqrt[(1+c^2*x^2)/(c^2*x^2)]/Sqrt[d+e*x^2]*Subst[Int[(a+b*x)^p/Sin[x]^(2*(q+1)),x],x,ArcCot[c*x]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && ILtQ[2*(q+1),0] && Not[IntegerQ[q]]
```

2. $\int \frac{a+b\,\text{ArcTan}[c x]}{d+e x^2}\,dx$

1: $\int \frac{\text{ArcTan}[c x]}{d+e x^2}\,dx$

Derivation: Algebraic expansion

- Basis: $\text{ArcTan}[z] = \frac{1}{2} i\,\text{Log}[1-i z] - \frac{1}{2} i\,\text{Log}[1+i z]$
- Basis: $\text{ArcCot}[z] = \frac{1}{2} i\,\text{Log}\left[1-\frac{i}{z}\right] - \frac{1}{2} i\,\text{Log}\left[1+\frac{i}{z}\right]$
- Rule:

$$\int \frac{\text{ArcTan}[c\,x]}{d+e\,x^2}\,dx \longrightarrow \frac{i}{2}\int \frac{\text{Log}[1-i\,c\,x]}{d+e\,x^2}\,dx - \frac{i}{2}\int \frac{\text{Log}[1+i\,c\,x]}{d+e\,x^2}\,dx$$

Program code:

```
Int[ArcTan[c_.*x_]/(d_.+e_.*x_^2),x_Symbol] :=
  I/2*Int[Log[1-I*c*x]/(d+e*x^2),x] - I/2*Int[Log[1+I*c*x]/(d+e*x^2),x] /;
FreeQ[{c,d,e},x]
```

```
Int[ArcCot[c_.*x_]/(d_.+e_.*x_^2),x_Symbol] :=
  I/2*Int[Log[1-I/(c*x)]/(d+e*x^2),x] - I/2*Int[Log[1+I/(c*x)]/(d+e*x^2),x] /;
FreeQ[{c,d,e},x]
```

2: $\int \frac{a+b\,\text{ArcTan}[c\,x]}{d+e\,x^2}\,dx$

Derivation: Algebraic expansion

Rule:

$$\int \frac{a+b\,\text{ArcTan}[c\,x]}{d+e\,x^2}\,dx \longrightarrow a\int \frac{1}{d+e\,x^2}\,dx + b\int \frac{\text{ArcTan}[c\,x]}{d+e\,x^2}\,dx$$

Program code:

```
Int[(a_+b_.*ArcTan[c_.*x_])/(d_.+e_.*x_^2),x_Symbol] :=
  a*Int[1/(d+e*x^2),x] + b*Int[ArcTan[c*x]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x]
```

```
Int[(a_+b_.*ArcCot[c_.*x_])/(d_.+e_.*x_^2),x_Symbol] :=
  a*Int[1/(d+e*x^2),x] + b*Int[ArcCot[c*x]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x]
```

3: $\int (d+e x^2)^q (a+b \text{ArcTan}[c x]) \, dx$ when $q \in \mathbb{Z} \lor q+\frac{1}{2} \in \mathbb{Z}^-$

Derivation: Integration by parts

- **Note: If $q \in \mathbb{Z}^+ \lor q+\frac{1}{2} \in \mathbb{Z}^-$, then $\int (d+e x^2)^q \, dx$ is expressible as an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.**
- **Rule: If $q \in \mathbb{Z} \lor q+\frac{1}{2} \in \mathbb{Z}^-$, let $u = \int (d+e x^2)^q \, dx$, then**

$$\int (d+e x^2)^q (a+b \text{ArcTan}[c x]) \, dx \longrightarrow u (a+b \text{ArcTan}[c x]) - b c \int \frac{u}{1+c^2 x^2} \, dx$$

Program code:

```
Int[(d_.+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^q,x]},
  Dist[a+b*ArcTan[c*x],u,x] - b*c*Int[u/(1+c^2*x^2),x]] /;
FreeQ[{a,b,c,d,e},x] && (IntegerQ[q] || ILtQ[q+1/2,0])
```

```
Int[(d_.+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(d+e*x^2)^q,x]},
  Dist[a+b*ArcCot[c*x],u,x] + b*c*Int[u/(1+c^2*x^2),x]] /;
FreeQ[{a,b,c,d,e},x] && (IntegerQ[q] || ILtQ[q+1/2,0])
```

4: $\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx$ when $q \in \mathbb{Z} \land p \in \mathbb{Z}^+$

Rule: If $q \in \mathbb{Z} \land p \in \mathbb{Z}^+$, then

$$\int (d+e x^2)^q (a+b \text{ArcTan}[c x])^p \, dx \longrightarrow \int (a+b \text{ArcTan}[c x])^p \, \text{ExpandIntegrand}\left[(d+e x^2)^q, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcTan[c*x])^p,(d+e*x^2)^q,x],x] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[q] && IGtQ[p,0]
```

```
Int[(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCot[c*x])^p,(d+e*x^2)^q,x],x] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[q] && IGtQ[p,0]
```

6. $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx$

1. $\int \frac{(f\,x)^m (a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$

1: $\int \frac{(f\,x)^m (a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $p>0 \wedge m>1$

Derivation: Algebraic expansion

Basis: $\frac{x^2}{d+e\,x^2} == \frac{1}{e} - \frac{d}{e\,(d+e\,x^2)}$

Rule: If $p>0 \wedge m>1$, then

$$\int \frac{(f\,x)^m (a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx \longrightarrow \frac{f^2}{e}\int (f\,x)^{m-2} (a+b\,\text{ArcTan}[c\,x])^p\,dx - \frac{d\,f^2}{e}\int \frac{(f\,x)^{m-2} (a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  f^2/e*Int[(f*x)^(m-2)*(a+b*ArcTan[c*x])^p,x] -
  d*f^2/e*Int[(f*x)^(m-2)*(a+b*ArcTan[c*x])^p/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[p,0] && GtQ[m,1]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  f^2/e*Int[(f*x)^(m-2)*(a+b*ArcCot[c*x])^p,x] -
  d*f^2/e*Int[(f*x)^(m-2)*(a+b*ArcCot[c*x])^p/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[p,0] && GtQ[m,1]
```

2: $\int \frac{(f x)^m (a + b \text{ArcTan}[c x])^p}{d + e x^2} dx$ when $p > 0 \land m < -1$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e x^2} = \frac{1}{d} - \frac{e x^2}{d (d+e x^2)}$

Rule: If $p > 0 \land m < -1$**, then**

$$\int \frac{(f x)^m (a + b \text{ArcTan}[c x])^p}{d + e x^2} dx \longrightarrow \frac{1}{d} \int (f x)^m (a + b \text{ArcTan}[c x])^p dx - \frac{e}{d f^2} \int \frac{(f x)^{m+2} (a + b \text{ArcTan}[c x])^p}{d + e x^2} dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  1/d*Int[(f*x)^m*(a+b*ArcTan[c*x])^p,x] -
  e/(d*f^2)*Int[(f*x)^(m+2)*(a+b*ArcTan[c*x])^p/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[p,0] && LtQ[m,-1]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  1/d*Int[(f*x)^m*(a+b*ArcCot[c*x])^p,x] -
  e/(d*f^2)*Int[(f*x)^(m+2)*(a+b*ArcCot[c*x])^p/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,f},x] && GtQ[p,0] && LtQ[m,-1]
```

3. $\int \frac{(f\,x)^m\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $e == c^2\,d$

1. $\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $e == c^2\,d$

1: $\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $e == c^2\,d \land p \in \mathbb{Z}^+$

Derivation: Algebraic expansion and power rule for integration

Basis: If $e == c^2\,d$**, then** $\frac{x}{d+e\,x^2} == -\frac{i\,c}{e\,(1+c^2\,x^2)} - \frac{1}{c\,d\,(i-c\,x)}$

Rule: If $e == c^2\,d \land p \in \mathbb{Z}^+$**, then**

$$\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx \longrightarrow -\frac{i\,(a+b\,\text{ArcTan}[c\,x])^{p+1}}{b\,e\,(p+1)} - \frac{1}{c\,d}\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{i-c\,x}\,dx$$

Program code:

```
Int[x_*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  -I*(a+b*ArcTan[c*x])^(p+1)/(b*e*(p+1)) -
  1/(c*d)*Int[(a+b*ArcTan[c*x])^p/(I-c*x),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0]
```

```
Int[x_*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  I*(a+b*ArcCot[c*x])^(p+1)/(b*e*(p+1)) -
  1/(c*d)*Int[(a+b*ArcCot[c*x])^p/(I-c*x),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0]
```

2: $\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $e == c^2\,d \land p \notin \mathbb{Z}^+ \land p \neq -1$

Derivation: Integration by parts

Basis: If $e == c^2\,d$, then $\frac{(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2} == \partial_x \frac{(a+b\,\text{ArcTan}[c\,x])^{p+1}}{b\,c\,d\,(p+1)}$

Rule: If $e == c^2\,d \land p \notin \mathbb{Z}^+ \land p \neq -1$, then

$$\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx \longrightarrow \frac{x\,(a+b\,\text{ArcTan}[c\,x])^{p+1}}{b\,c\,d\,(p+1)} - \frac{1}{b\,c\,d\,(p+1)}\int (a+b\,\text{ArcTan}[c\,x])^{p+1}\,dx$$

Program code:

```
Int[x_*(a_.+b_.*ArcTan[c_.*x_])^p_/(d_+e_.*x_^2),x_Symbol] :=
  x*(a+b*ArcTan[c*x])^(p+1)/(b*c*d*(p+1)) -
  1/(b*c*d*(p+1))*Int[(a+b*ArcTan[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && Not[IGtQ[p,0]] && NeQ[p,-1]
```

```
Int[x_*(a_.+b_.*ArcCot[c_.*x_])^p_/(d_+e_.*x_^2),x_Symbol] :=
  -x*(a+b*ArcCot[c*x])^(p+1)/(b*c*d*(p+1)) +
  1/(b*c*d*(p+1))*Int[(a+b*ArcCot[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && Not[IGtQ[p,0]] && NeQ[p,-1]
```

2: $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\,(d+e\,x^2)}\,dx$ when $e == c^2\,d \land p>0$

Derivation: Algebraic expansion

Basis: If $e == c^2\,d$, then $\frac{1}{x\,(d+e\,x^2)} == -\frac{i\,c}{d+e\,x^2}+\frac{i}{d\,x\,(i+c\,x)}$

Rule: If $e == c^2\,d \land p>0$, then

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\,(d+e\,x^2)}\,dx \longrightarrow -\frac{i\,(a+b\,\text{ArcTan}[c\,x])^{p+1}}{b\,d\,(p+1)}+\frac{i}{d}\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\,(i+c\,x)}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./(x_*(d_+e_.*x_^2)),x_Symbol] :=
  -I*(a+b*ArcTan[c*x])^(p+1)/(b*d*(p+1)) +
  I/d*Int[(a+b*ArcTan[c*x])^p/(x*(I+c*x)),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./(x_*(d_+e_.*x_^2)),x_Symbol] :=
  I*(a+b*ArcCot[c*x])^(p+1)/(b*d*(p+1)) +
  I/d*Int[(a+b*ArcCot[c*x])^p/(x*(I+c*x)),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,0]
```

3: $\int \frac{(f x)^m (a + b \operatorname{ArcTan}[c x])^p}{d + e x^2} \, dx$ when $e == c^2 d \land p < -1$

Derivation: Integration by parts

Basis: If $e == c^2 d$**, then** $\frac{(a+b \operatorname{ArcTan}[c x])^p}{d+e x^2} == \partial_x \frac{(a+b \operatorname{ArcTan}[c x])^{p+1}}{b c d (p+1)}$

Rule: If $e == c^2 d \land p < -1$**, then**

$$\int \frac{(f x)^m (a + b \operatorname{ArcTan}[c x])^p}{d + e x^2} \, dx \longrightarrow \frac{(f x)^m (a + b \operatorname{ArcTan}[c x])^{p+1}}{b c d (p + 1)} - \frac{f m}{b c d (p + 1)} \int (f x)^{m-1} (a + b \operatorname{ArcTan}[c x])^{p+1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcTan[c_.*x_])^p_/(d_+e_.*x_^2),x_Symbol] :=
  (f*x)^m*(a+b*ArcTan[c*x])^(p+1)/(b*c*d*(p+1)) -
  f*m/(b*c*d*(p+1))*Int[(f*x)^(m-1)*(a+b*ArcTan[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && LtQ[p,-1]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCot[c_.*x_])^p_/(d_+e_.*x_^2),x_Symbol] :=
  -(f*x)^m*(a+b*ArcCot[c*x])^(p+1)/(b*c*d*(p+1)) +
  f*m/(b*c*d*(p+1))*Int[(f*x)^(m-1)*(a+b*ArcCot[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && LtQ[p,-1]
```

4: $\int \frac{x^m (a + b \operatorname{ArcTan}[c x])}{d + e x^2} \, dx$ when $m \in \mathbb{Z} \land \neg (m == 1 \land a \neq 0)$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z} \land \neg (m == 1 \land a \neq 0)$**, then**

$$\int \frac{x^m (a + b \operatorname{ArcTan}[c x])}{d + e x^2} \, dx \longrightarrow \int (a + b \operatorname{ArcTan}[c x]) \operatorname{ExpandIntegrand}\left[\frac{x^m}{d + e x^2}, x\right] dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcTan[c_.*x_])/(d_+e_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcTan[c*x]),x^m/(d+e*x^2),x],x] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m] && Not[EqQ[m,1] && NeQ[a,0]]
```

```
Int[x_^m_.*(a_.+b_.*ArcCot[c_.*x_])/(d_+e_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCot[c*x]),x^m/(d+e*x^2),x],x] /;
FreeQ[{a,b,c,d,e},x] && IntegerQ[m] && Not[EqQ[m,1] && NeQ[a,0]]
```

2. $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $e == c^2 d$

   1. $\int x \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $e == c^2 d$

      1: $\int x \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $e == c^2 d \land p > 0 \land q \neq -1$

Derivation: Integration by parts

Rule: If $e == c^2 d \land p > 0 \land q \neq -1$, then

$$\int x \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx \longrightarrow \frac{\left(d+e\,x^2\right)^{q+1} (a+b\,\text{ArcTan}[c\,x])^p}{2\,e\,(q+1)} - \frac{b\,p}{2\,c\,(q+1)} \int \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^{p-1}\,dx$$

Program code:

```
Int[x_*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  (d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^p/(2*e*(q+1)) -
  b*p/(2*c*(q+1))*Int[(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p-1),x] /;
FreeQ[{a,b,c,d,e,q},x] && EqQ[e,c^2*d] && GtQ[p,0] && NeQ[q,-1]
```

```
Int[x_*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  (d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^p/(2*e*(q+1)) +
  b*p/(2*c*(q+1))*Int[(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p-1),x] /;
FreeQ[{a,b,c,d,e,q},x] && EqQ[e,c^2*d] && GtQ[p,0] && NeQ[q,-1]
```

2: $\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{(d+e\,x^2)^2}\,dx$ when $e == c^2 d \wedge p < -1 \wedge p \neq -2$

Rule: If $e == c^2 d \wedge p < -1 \wedge p \neq -2$, then

$$\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{(d+e\,x^2)^2}\,dx \longrightarrow \frac{x\,(a+b\,\text{ArcTan}[c\,x])^{p+1}}{b\,c\,d\,(p+1)\,(d+e\,x^2)} - \frac{(1-c^2\,x^2)\,(a+b\,\text{ArcTan}[c\,x])^{p+2}}{b^2\,e\,(p+1)\,(p+2)\,(d+e\,x^2)} - \frac{4}{b^2\,(p+1)\,(p+2)}\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^{p+2}}{(d+e\,x^2)^2}\,dx$$

Program code:

```
Int[x_*(a_.+b_.*ArcTan[c_.*x_])^p_/(d_+e_.*x_^2)^2,x_Symbol] :=
  x*(a+b*ArcTan[c*x])^(p+1)/(b*c*d*(p+1)*(d+e*x^2)) -
  (1-c^2*x^2)*(a+b*ArcTan[c*x])^(p+2)/(b^2*e*(p+1)*(p+2)*(d+e*x^2)) -
  4/(b^2*(p+1)*(p+2))*Int[x*(a+b*ArcTan[c*x])^(p+2)/(d+e*x^2)^2,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[p,-1] && NeQ[p,-2]
```

```
Int[x_*(a_.+b_.*ArcCot[c_.*x_])^p_/(d_+e_.*x_^2)^2,x_Symbol] :=
  -x*(a+b*ArcCot[c*x])^(p+1)/(b*c*d*(p+1)*(d+e*x^2)) -
  (1-c^2*x^2)*(a+b*ArcCot[c*x])^(p+2)/(b^2*e*(p+1)*(p+2)*(d+e*x^2)) -
  4/(b^2*(p+1)*(p+2))*Int[x*(a+b*ArcCot[c*x])^(p+2)/(d+e*x^2)^2,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[p,-1] && NeQ[p,-2]
```

2. $\int x^2 \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d$

1: $\int x^2 \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x]) \, dx$ when $e == c^2 d \land q < -1$

- **Rule: If $q == -\frac{5}{2}$, then better to use rule for when $m + 2 q + 3 == 0$.**

- **Rule: If $e == c^2 d \land q < -1$, then**

$$\int x^2 \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x]) \, dx \longrightarrow -\frac{b \left(d+e x^2\right)^{q+1}}{4 c^3 d (q+1)^2}+\frac{x \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])}{2 c^2 d (q+1)}-\frac{1}{2 c^2 d (q+1)} \int \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x]) \, dx$$

Program code:

```
Int[x_^2*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  -b*(d+e*x^2)^(q+1)/(4*c^3*d*(q+1)^2) +
  x*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])/(2*c^2*d*(q+1)) -
  1/(2*c^2*d*(q+1))*Int[(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[q,-1] && NeQ[q,-5/2]
```

```
Int[x_^2*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  b*(d+e*x^2)^(q+1)/(4*c^3*d*(q+1)^2) +
  x*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])/(2*c^2*d*(q+1)) -
  1/(2*c^2*d*(q+1))*Int[(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && LtQ[q,-1] && NeQ[q,-5/2]
```

2: $\int \frac{x^2 (a + b \text{ArcTan}[c x])^p}{(d + e x^2)^2} dx$ when $e == c^2 d \wedge p > 0$

Rule: If $e == c^2 d \wedge p > 0$, then

$$\int \frac{x^2 (a + b \text{ArcTan}[c x])^p}{(d + e x^2)^2} dx \longrightarrow \frac{(a + b \text{ArcTan}[c x])^{p+1}}{2 b c^3 d^2 (p + 1)} - \frac{x (a + b \text{ArcTan}[c x])^p}{2 c^2 d (d + e x^2)} + \frac{b p}{2 c} \int \frac{x (a + b \text{ArcTan}[c x])^{p-1}}{(d + e x^2)^2} dx$$

Program code:

```
Int[x_^2*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2)^2,x_Symbol] :=
  (a+b*ArcTan[c*x])^(p+1)/(2*b*c^3*d^2*(p+1)) -
  x*(a+b*ArcTan[c*x])^p/(2*c^2*d*(d+e*x^2)) +
  b*p/(2*c)*Int[x*(a+b*ArcTan[c*x])^(p-1)/(d+e*x^2)^2,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,0]
```

```
Int[x_^2*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2)^2,x_Symbol] :=
  -(a+b*ArcCot[c*x])^(p+1)/(2*b*c^3*d^2*(p+1)) -
  x*(a+b*ArcCot[c*x])^p/(2*c^2*d*(d+e*x^2)) -
  b*p/(2*c)*Int[x*(a+b*ArcCot[c*x])^(p-1)/(d+e*x^2)^2,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,0]
```

3. $\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e = c^2 d \land m+2q+2 = 0$

1. $\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e = c^2 d \land m+2q+2 = 0 \land q < -1 \land p \geq 1$

1: $\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x]) \, dx$ when $e = c^2 d \land m+2q+2 = 0 \land q < -1$

- **Rule: If $e = c^2 d \land m+2q+2 = 0 \land q < -1$, then**

$$\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x]) \, dx \longrightarrow$$

$$\frac{b (f x)^m \left(d+e x^2\right)^{q+1}}{c d m^2} - \frac{f (f x)^{m-1} \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])}{c^2 d m} + \frac{f^2 (m-1)}{c^2 d m} \int (f x)^{m-2} \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x]) \, dx$$

- **Program code:**

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  b*(f*x)^m*(d+e*x^2)^(q+1)/(c*d*m^2) -
  f*(f*x)^(m-1)*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])/(c^2*d*m) +
  f^2*(m-1)/(c^2*d*m)*Int[(f*x)^(m-2)*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && EqQ[m+2*q+2,0] && LtQ[q,-1]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  -b*(f*x)^m*(d+e*x^2)^(q+1)/(c*d*m^2) -
  f*(f*x)^(m-1)*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])/(c^2*d*m) +
  f^2*(m-1)/(c^2*d*m)*Int[(f*x)^(m-2)*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && EqQ[m+2*q+2,0] && LtQ[q,-1]
```

2: $\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e = c^2 d \land m+2q+2 = 0 \land q < -1 \land p > 1$

- **Rule: If $e = c^2 d \land m+2q+2 = 0 \land q < -1 \land p > 1$, then**

$$\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx \longrightarrow$$

$$\frac{b p (f x)^m \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])^{p-1}}{c d m^2} - \frac{f (f x)^{m-1} \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])^p}{c^2 d m} -$$

$$\frac{b^2 p (p-1)}{m^2} \int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^{p-2} \, dx+\frac{f^2 (m-1)}{c^2 d m} \int (f x)^{m-2} \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])^p \, dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_,x_Symbol] :=
  b*p*(f*x)^m*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^(p-1)/(c*d*m^2) -
  f*(f*x)^(m-1)*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^p/(c^2*d*m) -
  b^2*p*(p-1)/m^2*Int[(f*x)^m*(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p-2),x] +
  f^2*(m-1)/(c^2*d*m)*Int[(f*x)^(m-2)*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && EqQ[m+2*q+2,0] && LtQ[q,-1] && GtQ[p,1]
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_,x_Symbol] :=
  -b*p*(f*x)^m*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^(p-1)/(c*d*m^2) -
  f*(f*x)^(m-1)*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^p/(c^2*d*m) -
  b^2*p*(p-1)/m^2*Int[(f*x)^m*(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p-2),x] +
  f^2*(m-1)/(c^2*d*m)*Int[(f*x)^(m-2)*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && EqQ[m+2*q+2,0] && LtQ[q,-1] && GtQ[p,1]
```

2: $\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \land m+2 q+2 == 0 \land p < -1$

Derivation: Integration by parts

- **Basis: If** $e == c^2 d$, **then** $\frac{(a+b \operatorname{ArcTan}[c x])^p}{d+e x^2} == \partial_x \frac{(a+b \operatorname{ArcTan}[c x])^{p+1}}{b c d (p+1)}$
- **Basis: If** $m+2 q+2 == 0$, **then** $\partial_x \left(x^m \left(d+e x^2\right)^{q+1}\right) == c m x^{m-1} \left(d+e x^2\right)^q$

Rule: If $e == c^2 d \land m+2 q+2 == 0 \land p < -1$, **then**

$$\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx \longrightarrow \frac{(f x)^m \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])^{p+1}}{b c d (p+1)}-\frac{f m}{b c (p+1)} \int (f x)^{m-1} \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^{p+1} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_,x_Symbol] :=
  (f*x)^m*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^(p+1)/(b*c*d*(p+1)) -
  f*m/(b*c*(p+1))*Int[(f*x)^(m-1)*(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && EqQ[e,c^2*d] && EqQ[m+2*q+2,0] && LtQ[p,-1]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_,x_Symbol] :=
  -(f*x)^m*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^(p+1)/(b*c*d*(p+1)) +
  f*m/(b*c*(p+1))*Int[(f*x)^(m-1)*(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && EqQ[e,c^2*d] && EqQ[m+2*q+2,0] && LtQ[p,-1]
```

4: $\int (f\,x)^m\,\left(d+e\,x^2\right)^q\,(a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $e == c^2\,d\,\land\,m+2\,q+3 == 0\,\land\,p>0\,\land\,m\neq -1$

- Derivation: Integration by parts
- Basis: If $m+2\,q+3 == 0$, then $x^m\,\left(d+e\,x^2\right)^q == \partial_x\,\frac{x^{m+1}\,(d+e\,x^2)^{q+1}}{d\,(m+1)}$
- Rule: If $e == c^2\,d\,\land\,m+2\,q+3 == 0\,\land\,p>0\,\land\,m\neq -1$, then

$$\int (f\,x)^m\,\left(d+e\,x^2\right)^q\,(a+b\,\text{ArcTan}[c\,x])^p\,dx \longrightarrow \frac{(f\,x)^{m+1}\,\left(d+e\,x^2\right)^{q+1}\,(a+b\,\text{ArcTan}[c\,x])^p}{d\,f\,(m+1)}-\frac{b\,c\,p}{f\,(m+1)}\int (f\,x)^{m+1}\,\left(d+e\,x^2\right)^q\,(a+b\,\text{ArcTan}[c\,x])^{p-1}\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^p/(d*f*(m+1)) -
  b*c*p/(f*(m+1))*Int[(f*x)^(m+1)*(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && EqQ[e,c^2*d] && EqQ[m+2*q+3,0] && GtQ[p,0] && NeQ[m,-1]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  (f*x)^(m+1)*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^p/(d*f*(m+1)) +
  b*c*p/(f*(m+1))*Int[(f*x)^(m+1)*(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p-1),x] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && EqQ[e,c^2*d] && EqQ[m+2*q+3,0] && GtQ[p,0] && NeQ[m,-1]
```

5. $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $e == c^2\,d \wedge q>0$

1: $\int (f\,x)^m \sqrt{d+e\,x^2}\,(a+b\,\text{ArcTan}[c\,x])\,dx$ when $e == c^2\,d \wedge m \neq -2$

Rule: If $e == c^2\,d \wedge m \neq -2$**, then**

$$\int (f\,x)^m \sqrt{d+e\,x^2}\,(a+b\,\text{ArcTan}[c\,x])\,dx \longrightarrow \frac{(f\,x)^{m+1}\sqrt{d+e\,x^2}\,(a+b\,\text{ArcTan}[c\,x])}{f\,(m+2)} - \frac{b\,c\,d}{f\,(m+2)}\int \frac{(f\,x)^{m+1}}{\sqrt{d+e\,x^2}}\,dx + \frac{d}{m+2}\int \frac{(f\,x)^m\,(a+b\,\text{ArcTan}[c\,x])}{\sqrt{d+e\,x^2}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcTan[c*x])/(f*(m+2)) -
  b*c*d/(f*(m+2))*Int[(f*x)^(m+1)/Sqrt[d+e*x^2],x] +
  d/(m+2)*Int[(f*x)^m*(a+b*ArcTan[c*x])/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && NeQ[m,-2]
```

```
Int[(f_.*x_)^m_*Sqrt[d_+e_.*x_^2]*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcCot[c*x])/(f*(m+2)) +
  b*c*d/(f*(m+2))*Int[(f*x)^(m+1)/Sqrt[d+e*x^2],x] +
  d/(m+2)*Int[(f*x)^m*(a+b*ArcCot[c*x])/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && NeQ[m,-2]
```

2: $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $e == c^2\,d \wedge p \in \mathbb{Z}^+ \wedge q-1 \in \mathbb{Z}^+$

Rule: If $e == c^2\,d \wedge p \in \mathbb{Z}^+ \wedge q-1 \in \mathbb{Z}^+$**, then**

$$\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p,\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^2)^q*(a+b*ArcTan[c*x])^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && IGtQ[p,0] && IGtQ[q,1] && (EqQ[p,1] || IntegerQ[m])
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(f*x)^m*(d+e*x^2)^q*(a+b*ArcCot[c*x])^p,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && IGtQ[p,0] && IGtQ[q,1] && (EqQ[p,1] || IntegerQ[m])
```

3: $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p \,dx$ when $e == c^2 d \land q>0 \land p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis: If** $e == c^2 d$, **then** $\left(d+e\,x^2\right)^q == d\left(d+e\,x^2\right)^{q-1} + c^2 d\,x^2\left(d+e\,x^2\right)^{q-1}$

Rule: If $e == c^2 d \land q>0 \land p \in \mathbb{Z}^+$, **then**

$$\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p \,dx \longrightarrow d\int (f\,x)^m \left(d+e\,x^2\right)^{q-1} (a+b\,\text{ArcTan}[c\,x])^p \,dx + \frac{c^2 d}{f^2}\int (f\,x)^{m+2} \left(d+e\,x^2\right)^{q-1} (a+b\,\text{ArcTan}[c\,x])^p \,dx$$

Program code:

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  d*Int[(f*x)^m*(d+e*x^2)^(q-1)*(a+b*ArcTan[c*x])^p,x] +
  c^2*d/f^2*Int[(f*x)^(m+2)*(d+e*x^2)^(q-1)*(a+b*ArcTan[c*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[q,0] && IGtQ[p,0] && (RationalQ[m] || EqQ[p,1] && IntegerQ[q])
```

```
Int[(f_.*x_)^m_*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  d*Int[(f*x)^m*(d+e*x^2)^(q-1)*(a+b*ArcCot[c*x])^p,x] +
  c^2*d/f^2*Int[(f*x)^(m+2)*(d+e*x^2)^(q-1)*(a+b*ArcCot[c*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,m},x] && EqQ[e,c^2*d] && GtQ[q,0] && IGtQ[p,0] && (RationalQ[m] || EqQ[p,1] && IntegerQ[q])
```

6. $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $e == c^2\,d \wedge q<0$

1. $\int \frac{(f\,x)^m (a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d$

1: $\int \frac{(f\,x)^m (a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \wedge p>0 \wedge m>1$

–

Rule: If $e == c^2\,d \wedge p>0 \wedge m>1$, then

$$\int \frac{(f\,x)^m (a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx \longrightarrow$$

$$\frac{f\,(f\,x)^{m-1}\sqrt{d+e\,x^2}\,(a+b\,\text{ArcTan}[c\,x])^p}{c^2\,d\,m}-\frac{b\,f\,p}{c\,m}\int \frac{(f\,x)^{m-1}(a+b\,\text{ArcTan}[c\,x])^{p-1}}{\sqrt{d+e\,x^2}}\,dx-\frac{f^2\,(m-1)}{c^2\,m}\int \frac{(f\,x)^{m-2}(a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcTan[c_.*x_])^p_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d+e*x^2]*(a+b*ArcTan[c*x])^p/(c^2*d*m) -
  b*f*p/(c*m)*Int[(f*x)^(m-1)*(a+b*ArcTan[c*x])^(p-1)/Sqrt[d+e*x^2],x] -
  f^2*(m-1)/(c^2*m)*Int[(f*x)^(m-2)*(a+b*ArcTan[c*x])^p/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[p,0] && GtQ[m,1]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCot[c_.*x_])^p_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  f*(f*x)^(m-1)*Sqrt[d+e*x^2]*(a+b*ArcCot[c*x])^p/(c^2*d*m) +
  b*f*p/(c*m)*Int[(f*x)^(m-1)*(a+b*ArcCot[c*x])^(p-1)/Sqrt[d+e*x^2],x] -
  f^2*(m-1)/(c^2*m)*Int[(f*x)^(m-2)*(a+b*ArcCot[c*x])^p/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[p,0] && GtQ[m,1]
```

2. $\int \frac{(f\,x)^m (a+b\,\text{ArcTan}[c\,x])^p}{\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \wedge p>0 \wedge m\le -1$

1. $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\,\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \wedge p\in\mathbb{Z}^+$

1. $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\,\sqrt{d+e\,x^2}}\,dx$ when $e == c^2\,d \wedge p\in\mathbb{Z}^+ \wedge d>0$

1: $\int \frac{(a + b \operatorname{ArcTan}[c\, x])}{x \sqrt{d + e\, x^2}}\, dx$ when $e == c^2 d \wedge d > 0$

- **Derivation: Integration by substitution, piecewise constant extraction and algebraic simplification!**

- **Note: Although not essential, these rules returns antiderivatives free of complex exponentials of the form** $e^{\operatorname{ArcTan}[c\, x]}$ **and** $i\, e^{\operatorname{ArcCot}[c\, x]}$.

- **Basis: If** $e == c^2 d \wedge d > 0$, **then** $\frac{1}{x \sqrt{d+e\, x^2}} == \frac{1}{\sqrt{d}} \operatorname{Csc}[\operatorname{ArcTan}[c\, x]]\, \partial_x \operatorname{ArcTan}[c\, x]$

- **Basis: If** $e == c^2 d \wedge d > 0$, **then** $\frac{1}{x \sqrt{d+e\, x^2}} == -\frac{1}{\sqrt{d}} \frac{\operatorname{Csc}[\operatorname{ArcCot}[c\, x]] \operatorname{Sec}[\operatorname{ArcCot}[c\, x]]}{\sqrt{\operatorname{Csc}[\operatorname{ArcCot}[c\, x]]^2}}\, \partial_x \operatorname{ArcCot}[c\, x]$

Rule: If $e == c^2 d \wedge d > 0$, **then**

$$\int \frac{(a + b \operatorname{ArcTan}[c\, x])}{x \sqrt{d + e\, x^2}}\, dx \longrightarrow \frac{1}{\sqrt{d}} \operatorname{Subst}\left[\int (a + b\, x) \operatorname{Csc}[x]\, dx,\ x,\ \operatorname{ArcTan}[c\, x]\right]$$

$$\longrightarrow -\frac{2}{\sqrt{d}} (a + b \operatorname{ArcTan}[c\, x]) \operatorname{ArcTanh}\left[\frac{\sqrt{1 + i\, c\, x}}{\sqrt{1 - i\, c\, x}}\right] + \frac{i\, b}{\sqrt{d}} \operatorname{PolyLog}\left[2, -\frac{\sqrt{1 + i\, c\, x}}{\sqrt{1 - i\, c\, x}}\right] - \frac{i\, b}{\sqrt{d}} \operatorname{PolyLog}\left[2, \frac{\sqrt{1 + i\, c\, x}}{\sqrt{1 - i\, c\, x}}\right]$$

- **Program code:**

```
Int[(a_.+b_.*ArcTan[c_.*x_])/(x_*Sqrt[d_+e_.*x_^2]),x_Symbol] :=
  -2/Sqrt[d]*(a+b*ArcTan[c*x])*ArcTanh[Sqrt[1+I*c*x]/Sqrt[1-I*c*x]] +
  I*b/Sqrt[d]*PolyLog[2,-Sqrt[1+I*c*x]/Sqrt[1-I*c*x]] -
  I*b/Sqrt[d]*PolyLog[2,Sqrt[1+I*c*x]/Sqrt[1-I*c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[d,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])/(x_*Sqrt[d_+e_.*x_^2]),x_Symbol] :=
  -2/Sqrt[d]*(a+b*ArcCot[c*x])*ArcTanh[Sqrt[1+I*c*x]/Sqrt[1-I*c*x]] -
  I*b/Sqrt[d]*PolyLog[2,-Sqrt[1+I*c*x]/Sqrt[1-I*c*x]] +
  I*b/Sqrt[d]*PolyLog[2,Sqrt[1+I*c*x]/Sqrt[1-I*c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[d,0]
```

2. $\int \frac{(a + b\,\text{ArcTan}[c\,x])^p}{x\,\sqrt{d + e\,x^2}}\,dx$ when $e == c^2\,d \land p \in \mathbb{Z}^+ \land d > 0$

1: $\int \frac{(a + b\,\text{ArcTan}[c\,x])^p}{x\,\sqrt{d + e\,x^2}}\,dx$ when $e == c^2\,d \land p \in \mathbb{Z}^+ \land d > 0$

Derivation: Integration by substitution

- **Basis: If** $e == c^2\,d \land d > 0$**, then** $\frac{1}{x\,\sqrt{d+e\,x^2}} == \frac{1}{\sqrt{d}}\,\text{Csc}[\text{ArcTan}[c\,x]]\,\partial_x\,\text{ArcTan}[c\,x]$

- **Rule: If** $e == c^2\,d \land p \in \mathbb{Z}^+ \land d > 0$**, then**

$$\int \frac{(a + b\,\text{ArcTan}[c\,x])^p}{x\,\sqrt{d + e\,x^2}}\,dx \longrightarrow \frac{1}{\sqrt{d}}\,\text{Subst}\Big[\int (a + b\,x)^p\,\text{Csc}[x]\,dx,\ x,\ \text{ArcTan}[c\,x]\Big]$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_/(x_*Sqrt[d_+e_.*x_^2]),x_Symbol] :=
  1/Sqrt[d]*Subst[Int[(a+b*x)^p*Csc[x],x],x,ArcTan[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && GtQ[d,0]
```

2: $\int \frac{(a + b\,\text{ArcCot}[c\,x])^p}{x\,\sqrt{d + e\,x^2}}\,dx$ when $e == c^2\,d \land p \in \mathbb{Z}^+ \land d > 0$

Derivation: Integration by substitution and piecewise constant extraction

- **Basis: If** $e == c^2\,d \land d > 0$**, then** $\frac{1}{x\,\sqrt{d+e\,x^2}} == -\frac{1}{\sqrt{d}}\,\frac{\text{Csc}[\text{ArcCot}[c\,x]]\,\text{Sec}[\text{ArcCot}[c\,x]]}{\sqrt{\text{Csc}[\text{ArcCot}[c\,x]]^2}}\,\partial_x\,\text{ArcCot}[c\,x]$

- **Basis:** $\partial_x\,\frac{\text{Csc}[x]}{\sqrt{\text{Csc}[x]^2}} == 0$

- **Basis:** $\frac{\text{Csc}[\text{ArcCot}[c\,x]]}{\sqrt{\text{Csc}[\text{ArcCot}[c\,x]]^2}} == \frac{c\,x\,\sqrt{1+\frac{1}{c^2\,x^2}}}{\sqrt{1+c^2\,x^2}}$

- **Rule: If** $e == c^2\,d \land p \in \mathbb{Z}^+ \land d > 0$**, then**

$$\int \frac{(a + b\,\text{ArcCot}[c\,x])^p}{x\,\sqrt{d + e\,x^2}}\,dx \longrightarrow -\frac{1}{\sqrt{d}}\,\text{Subst}\Big[\int \frac{(a + b\,x)^p\,\text{Csc}[x]\,\text{Sec}[x]}{\sqrt{\text{Csc}[x]^2}}\,dx,\ x,\ \text{ArcCot}[c\,x]\Big]$$

$$\rightarrow -\frac{c\,x\sqrt{1+\frac{1}{c^2 x^2}}}{\sqrt{d+e\,x^2}}\,\text{Subst}\left[\int (a+b\,x)^p\,\text{Sec}[x]\,dx,\,x,\,\text{ArcCot}[c\,x]\right]$$

Program code:

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_/(x_*Sqrt[d_+e_.*x_^2]),x_Symbol] :=
  -c*x*Sqrt[1+1/(c^2*x^2)]/Sqrt[d+e*x^2]*Subst[Int[(a+b*x)^p*Sec[x],x],x,ArcCot[c*x]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && GtQ[d,0]
```

2: $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\sqrt{d+e\,x^2}}\,dx$ when $e == c^2 d \land p \in \mathbb{Z}^+ \land d \not> 0$

Derivation: Piecewise constant extraction

- **Basis: If** $e == c^2 d$**, then** $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e\,x^2}} == 0$

Rule: If $e == c^2 d \land p \in \mathbb{Z}^+ \land d \not> 0$**, then**

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\sqrt{d+e\,x^2}}\,dx \rightarrow \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e\,x^2}}\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{x\sqrt{1+c^2 x^2}}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./(x_*Sqrt[d_+e_.*x_^2]),x_Symbol] :=
  Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[(a+b*ArcTan[c*x])^p/(x*Sqrt[1+c^2*x^2]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && Not[GtQ[d,0]]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./(x_*Sqrt[d_+e_.*x_^2]),x_Symbol] :=
  Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[(a+b*ArcCot[c*x])^p/(x*Sqrt[1+c^2*x^2]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && Not[GtQ[d,0]]
```

2. $\int \frac{(f x)^m (a + b \operatorname{ArcTan}[c x])^p}{\sqrt{d + e x^2}} \, dx$ when $e == c^2 d \land p > 0 \land m < -1$

1: $\int \frac{(a + b \operatorname{ArcTan}[c x])^p}{x^2 \sqrt{d + e x^2}} \, dx$ when $e == c^2 d \land p > 0$

Derivation: Integration by parts

- **Basis:** $\frac{1}{x^2 \sqrt{d+e x^2}} == -\partial_x \frac{\sqrt{d+e x^2}}{d x}$

Rule: If $e == c^2 d \land p > 0$, then

$$\int \frac{(a + b \operatorname{ArcTan}[c x])^p}{x^2 \sqrt{d + e x^2}} \, dx \longrightarrow -\frac{\sqrt{d + e x^2} \, (a + b \operatorname{ArcTan}[c x])^p}{d x} + b c p \int \frac{(a + b \operatorname{ArcTan}[c x])^{p-1}}{x \sqrt{d + e x^2}} \, dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_./(x_^2*Sqrt[d_+e_.*x_^2]),x_Symbol] :=
  -Sqrt[d+e*x^2]*(a+b*ArcTan[c*x])^p/(d*x) +
  b*c*p*Int[(a+b*ArcTan[c*x])^(p-1)/(x*Sqrt[d+e*x^2]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_./(x_^2*Sqrt[d_+e_.*x_^2]),x_Symbol] :=
  -Sqrt[d+e*x^2]*(a+b*ArcCot[c*x])^p/(d*x) -
  b*c*p*Int[(a+b*ArcCot[c*x])^(p-1)/(x*Sqrt[d+e*x^2]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && GtQ[p,0]
```

2: $\int \frac{(f x)^m (a + b \operatorname{ArcTan}[c x])^p}{\sqrt{d + e x^2}} dx$ when $e == c^2 d \wedge p > 0 \wedge m < -1 \wedge m \neq -2$

Rule: If $e == c^2 d \wedge p > 0 \wedge m < -1 \wedge m \neq -2$, then

$$\int \frac{(f x)^m (a + b \operatorname{ArcTan}[c x])^p}{\sqrt{d + e x^2}} dx \longrightarrow$$

$$\frac{(f x)^{m+1} \sqrt{d + e x^2} (a + b \operatorname{ArcTan}[c x])^p}{d f (m + 1)} - \frac{b c p}{f (m + 1)} \int \frac{(f x)^{m+1} (a + b \operatorname{ArcTan}[c x])^{p-1}}{\sqrt{d + e x^2}} dx - \frac{c^2 (m + 2)}{f^2 (m + 1)} \int \frac{(f x)^{m+2} (a + b \operatorname{ArcTan}[c x])^p}{\sqrt{d + e x^2}} dx$$

Program code:

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcTan[c_.*x_])^p_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcTan[c*x])^p/(d*f*(m+1)) -
  b*c*p/(f*(m+1))*Int[(f*x)^(m+1)*(a+b*ArcTan[c*x])^(p-1)/Sqrt[d+e*x^2],x] -
  c^2*(m+2)/(f^2*(m+1))*Int[(f*x)^(m+2)*(a+b*ArcTan[c*x])^p/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[p,0] && LtQ[m,-1] && NeQ[m,-2]
```

```
Int[(f_.*x_)^m_*(a_.+b_.*ArcCot[c_.*x_])^p_./Sqrt[d_+e_.*x_^2],x_Symbol] :=
  (f*x)^(m+1)*Sqrt[d+e*x^2]*(a+b*ArcCot[c*x])^p/(d*f*(m+1)) +
  b*c*p/(f*(m+1))*Int[(f*x)^(m+1)*(a+b*ArcCot[c*x])^(p-1)/Sqrt[d+e*x^2],x] -
  c^2*(m+2)/(f^2*(m+1))*Int[(f*x)^(m+2)*(a+b*ArcCot[c*x])^p/Sqrt[d+e*x^2],x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e,c^2*d] && GtQ[p,0] && LtQ[m,-1] && NeQ[m,-2]
```

2. $\int x^m \left(d + e x^2\right)^q (a + b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \wedge q < -1$

1: $\int x^m \left(d + e x^2\right)^q (a + b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \wedge (m \mid p \mid 2 q) \in \mathbb{Z} \wedge q < -1 \wedge m > 1 \wedge p \neq -1$

Derivation: Algebraic expansion

Basis: $\frac{x^2}{d+e x^2} == \frac{1}{e} - \frac{d}{e (d+e x^2)}$

Rule: If $e == c^2 d \wedge (m \mid p \mid 2 q) \in \mathbb{Z} \wedge q < -1 \wedge m > 1 \wedge p \neq -1$**, then**

$$\int x^m \left(d + e x^2\right)^q (a + b \operatorname{ArcTan}[c x])^p \, dx \longrightarrow \frac{1}{e} \int x^{m-2} \left(d + e x^2\right)^{q+1} (a + b \operatorname{ArcTan}[c x])^p \, dx - \frac{d}{e} \int x^{m-2} \left(d + e x^2\right)^q (a + b \operatorname{ArcTan}[c x])^p \, dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  1/e*Int[x^(m-2)*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^p,x] -
  d/e*Int[x^(m-2)*(d+e*x^2)^q*(a+b*ArcTan[c*x])^p,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IntegersQ[p,2*q] && LtQ[q,-1] && IGtQ[m,1] && NeQ[p,-1]
```

```
Int[x_^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  1/e*Int[x^(m-2)*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^p,x] -
  d/e*Int[x^(m-2)*(d+e*x^2)^q*(a+b*ArcCot[c*x])^p,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IntegersQ[p,2*q] && LtQ[q,-1] && IGtQ[m,1] && NeQ[p,-1]
```

2: $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \land (m \mid p \mid 2 q) \in \mathbb{Z} \land q < -1 \land m < 0 \land p \neq -1$

Derivation: Algebraic expansion

- **Basis:** $\frac{1}{d+e x^2} == \frac{1}{d} - \frac{e x^2}{d \left(d+e x^2\right)}$

- **Rule: If** $e == c^2 d \land (m \mid p \mid 2 q) \in \mathbb{Z} \land q < -1 \land m < 0 \land p \neq -1$**, then**

$$\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx \longrightarrow \frac{1}{d} \int x^m \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])^p \, dx - \frac{e}{d} \int x^{m+2} \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$$

Program code:

```
Int[x_^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  1/d*Int[x^m*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^p,x] -
  e/d*Int[x^(m+2)*(d+e*x^2)^q*(a+b*ArcTan[c*x])^p,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IntegersQ[p,2*q] && LtQ[q,-1] && ILtQ[m,0] && NeQ[p,-1]
```

```
Int[x_^m_*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  1/d*Int[x^m*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^p,x] -
  e/d*Int[x^(m+2)*(d+e*x^2)^q*(a+b*ArcCot[c*x])^p,x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IntegersQ[p,2*q] && LtQ[q,-1] && ILtQ[m,0] && NeQ[p,-1]
```

3: $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \land m \in \mathbb{Z} \land q < -1 \land p < -1 \land m+2q+2 \neq 0$

Derivation: Integration by parts

Rule: If $e == c^2 d \land m \in \mathbb{Z} \land q < -1 \land p < -1 \land m+2q+2 \neq 0$**, then**

$$\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx \longrightarrow$$

$$\frac{x^m \left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])^{p+1}}{b c d (p+1)} -$$

$$\frac{m}{b c (p+1)} \int x^{m-1} \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^{p+1} dx - \frac{c (m+2q+2)}{b (p+1)} \int x^{m+1} \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^{p+1} dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  x^m*(d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])^(p+1)/(b*c*d*(p+1)) -
  m/(b*c*(p+1))*Int[x^(m-1)*(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p+1),x] -
  c*(m+2*q+2)/(b*(p+1))*Int[x^(m+1)*(d+e*x^2)^q*(a+b*ArcTan[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[e,c^2*d] && IntegerQ[m] && LtQ[q,-1] && LtQ[p,-1] && NeQ[m+2*q+2,0]
```

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  -x^m*(d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])^(p+1)/(b*c*d*(p+1)) +
  m/(b*c*(p+1))*Int[x^(m-1)*(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p+1),x] +
  c*(m+2*q+2)/(b*(p+1))*Int[x^(m+1)*(d+e*x^2)^q*(a+b*ArcCot[c*x])^(p+1),x] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[e,c^2*d] && IntegerQ[m] && LtQ[q,-1] && LtQ[p,-1] && NeQ[m+2*q+2,0]
```

4. $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \land m \in \mathbb{Z}^+ \land m+2q+1 \in \mathbb{Z}^-$

1. $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \land m \in \mathbb{Z}^+ \land m+2q+1 \in \mathbb{Z}^-$

1: $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \land m \in \mathbb{Z}^+ \land m+2q+1 \in \mathbb{Z}^- \land (q \in \mathbb{Z} \lor d > 0)$

Derivation: Integration by substitution

Basis: If $e == c^2 d \land m \in \mathbb{Z} \land m+2q+1 \in \mathbb{Z} \land (q \in \mathbb{Z} \lor d > 0)$**, then** $x^m \left(d+e x^2\right)^q == \frac{d^q \operatorname{Sin}[\operatorname{ArcTan}[c x]]^m}{c^{m+1} \operatorname{Cos}[\operatorname{ArcTan}[c x]]^{m+2(q+1)}} \partial_x \operatorname{ArcTan}[c x]$

Rule: If $e == c^2 d \land m \in \mathbb{Z}^+ \land m+2q+1 \in \mathbb{Z}^- \land (q \in \mathbb{Z} \lor d > 0)$**, then**

$$\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx \longrightarrow \frac{d^q}{c^{m+1}} \operatorname{Subst}\left[\int \frac{(a+b x)^p \operatorname{Sin}[x]^m}{\operatorname{Cos}[x]^{m+2 (q+1)}} \, dx, x, \operatorname{ArcTan}[c x]\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  d^q/c^(m+1)*Subst[Int[(a+b*x)^p*Sin[x]^m/Cos[x]^(m+2*(q+1)),x],x,ArcTan[c*x]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && IGtQ[m,0] && ILtQ[m+2*q+1,0] && (IntegerQ[q] || GtQ[d,0])
```

2: $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$ when $e == c^2 d \land m \in \mathbb{Z}^+ \land m+2q+1 \in \mathbb{Z}^- \land \neg (q \in \mathbb{Z} \lor d > 0)$

Derivation: Piecewise constant extraction

- **Basis: If** $e == c^2 d$**, then** $\partial_x \frac{\sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} == 0$

Rule: If $e == c^2 d \land m \in \mathbb{Z}^+ \land m+2q+1 \in \mathbb{Z}^- \land \neg (q \in \mathbb{Z} \lor d > 0)$**, then**

$$\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx \longrightarrow \frac{d^{q+\frac{1}{2}} \sqrt{1+c^2 x^2}}{\sqrt{d+e x^2}} \int x^m \left(1+c^2 x^2\right)^q (a+b \operatorname{ArcTan}[c x])^p \, dx$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  d^(q+1/2)*Sqrt[1+c^2*x^2]/Sqrt[d+e*x^2]*Int[x^m*(1+c^2*x^2)^q*(a+b*ArcTan[c*x])^p,x] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && IGtQ[m,0] && ILtQ[m+2*q+1,0] && Not[IntegerQ[q] || GtQ[d,0]]
```

2. $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcCot}[c x])^p \,dx$ when $e == c^2 d \wedge m \in \mathbb{Z}^+ \wedge m+2q+1 \in \mathbb{Z}^-$

1: $\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcCot}[c x])^p \,dx$ when $e == c^2 d \wedge m \in \mathbb{Z}^+ \wedge m+2q+1 \in \mathbb{Z}^- \wedge q \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $e == c^2 d \wedge m \in \mathbb{Z} \wedge q \in \mathbb{Z}$**, then** $x^m \left(d+e x^2\right)^q == -\frac{d^q \operatorname{Cos}[\operatorname{ArcCot}[c x]]^m}{c^{m+1} \operatorname{Sin}[\operatorname{ArcCot}[c x]]^{m+2(q+1)}} \partial_x \operatorname{ArcCot}[c x]$

Rule: If $e == c^2 d \wedge m \in \mathbb{Z}^+ \wedge m+2q+1 \in \mathbb{Z}^- \wedge q \in \mathbb{Z}$**, then**

$$\int x^m \left(d+e x^2\right)^q (a+b \operatorname{ArcCot}[c x])^p \,dx \longrightarrow -\frac{d^q}{c^{m+1}} \operatorname{Subst}\left[\int \frac{(a+b x)^p \operatorname{Cos}[x]^m}{\operatorname{Sin}[x]^{m+2(q+1)}} \,dx,\ x,\ \operatorname{ArcCot}[c x]\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  -d^q/c^(m+1)*Subst[Int[(a+b*x)^p*Cos[x]^m/Sin[x]^(m+2*(q+1)),x],x,ArcCot[c*x]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && IGtQ[m,0] && ILtQ[m+2*q+1,0] && IntegerQ[q]
```

2: $\int x^m \left(d+e\,x^2\right)^q \,(a+b\,\text{ArcCot}[c\,x])^p\,dx$ when $e = c^2 d \wedge m \in \mathbb{Z}^+ \wedge m+2q+1 \in \mathbb{Z}^- \wedge q \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

- Basis: If $e = c^2 d$, then $\partial_x \frac{x\sqrt{\frac{1+c^2x^2}{c^2x^2}}}{\sqrt{d+e\,x^2}} = 0$

- Basis: If $m \in \mathbb{Z} \wedge m+2q+1 \in \mathbb{Z} \wedge q \notin \mathbb{Z}$, then $x^{m+1}\sqrt{1+\frac{1}{c^2x^2}}\left(1+c^2x^2\right)^{q-\frac{1}{2}} = -\frac{\text{Cos}[\text{ArcCot}[c\,x]]^m}{c^{m+2}\,\text{Sin}[\text{ArcCot}[c\,x]]^{m+2\,(q+1)}}\,\partial_x\,\text{ArcCot}[c\,x]$

Rule: If $e = c^2 d \wedge m \in \mathbb{Z}^+ \wedge m+2q+1 \in \mathbb{Z}^- \wedge q \notin \mathbb{Z}$, then

$$\int x^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcCot}[c\,x])^p\,dx \to \frac{c^2\,d^{q+\frac{1}{2}}\,x\sqrt{\frac{1+c^2x^2}{c^2x^2}}}{\sqrt{d+e\,x^2}}\int x^{m+1}\sqrt{1+\frac{1}{c^2x^2}}\left(1+c^2x^2\right)^{q-\frac{1}{2}}(a+b\,\text{ArcCot}[c\,x])^p\,dx$$

$$\to -\frac{d^{q+\frac{1}{2}}\,x\sqrt{\frac{1+c^2x^2}{c^2x^2}}}{c^m\sqrt{d+e\,x^2}}\,\text{Subst}\left[\int\frac{(a+b\,x)^p\,\text{Cos}[x]^m}{\text{Sin}[x]^{m+2\,(q+1)}}\,dx,\,x,\,\text{ArcCot}[c\,x]\right]$$

Program code:

```
Int[x_^m_.*(d_+e_.*x_^2)^q_*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  -d^(q+1/2)*x*Sqrt[(1+c^2*x^2)/(c^2*x^2)]/(c^m*Sqrt[d+e*x^2])*Subst[Int[(a+b*x)^p*Cos[x]^m/Sin[x]^(m+2*(q+1)),x],x,ArcCot[c*x]] /;
FreeQ[{a,b,c,d,e,p},x] && EqQ[e,c^2*d] && IGtQ[m,0] && ILtQ[m+2*q+1,0] && Not[IntegerQ[q]]
```

3. $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])\,dx$ when

$\left(q \in \mathbb{Z}^+ \bigwedge \neg\left(\frac{m-1}{2} \in \mathbb{Z}^- \bigwedge m+2q+3>0\right)\right) \bigvee \left(\frac{m+1}{2} \in \mathbb{Z}^+ \bigwedge \neg\,(q \in \mathbb{Z}^- \wedge m+2q+3>0)\right) \bigvee \left(\frac{m+2q+1}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \notin \mathbb{Z}^-\right)$

1: $\int x \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])\,dx$ when $q \neq -1$

Derivation: Integration by parts

- Basis: $x\left(d+e\,x^2\right)^q = \partial_x \frac{\left(d+e\,x^2\right)^{q+1}}{2\,e\,(q+1)}$

Rule: If $q \neq -1$, then

$$\int x \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x]) \, dx \longrightarrow \frac{\left(d+e x^2\right)^{q+1} (a+b \operatorname{ArcTan}[c x])}{2 e (q+1)} - \frac{b c}{2 e (q+1)} \int \frac{\left(d+e x^2\right)^{q+1}}{1+c^2 x^2} \, dx$$

Program code:

```
Int[x_*(d_.+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(q+1)*(a+b*ArcTan[c*x])/(2*e*(q+1)) -
  b*c/(2*e*(q+1))*Int[(d+e*x^2)^(q+1)/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[q,-1]
```

```
Int[x_*(d_.+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  (d+e*x^2)^(q+1)*(a+b*ArcCot[c*x])/(2*e*(q+1)) +
  b*c/(2*e*(q+1))*Int[(d+e*x^2)^(q+1)/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[q,-1]
```

2: $\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x]) \, dx$ when
$\left(q \in \mathbb{Z}^+ \bigwedge \neg \left(\frac{m-1}{2} \in \mathbb{Z}^- \bigwedge m+2q+3>0\right)\right) \bigvee \left(\frac{m+1}{2} \in \mathbb{Z}^+ \bigwedge \neg (q \in \mathbb{Z}^- \land m+2q+3>0)\right) \bigvee \left(\frac{m+2q+1}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \notin \mathbb{Z}^-\right)$

Derivation: Integration by parts

- **Note: If** $\left(q \in \mathbb{Z}^+ \bigwedge \neg \left(\frac{m-1}{2} \in \mathbb{Z}^- \bigwedge m+2q+3>0\right)\right) \bigvee \left(\frac{m+1}{2} \in \mathbb{Z}^+ \bigwedge \neg (q \in \mathbb{Z}^- \land m+2q+3>0)\right) \bigvee \left(\frac{m+2q+1}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \notin \mathbb{Z}^-\right)$, **then** $\int (f x)^m \left(d+e x^2\right)^q dx$ **is expressible as an algebraic function not involving logarithms, inverse trig or inverse hyperbolic functions.**
- **Rule: If** $\left(q \in \mathbb{Z}^+ \bigwedge \neg \left(\frac{m-1}{2} \in \mathbb{Z}^- \bigwedge m+2q+3>0\right)\right) \bigvee \left(\frac{m+1}{2} \in \mathbb{Z}^+ \bigwedge \neg (q \in \mathbb{Z}^- \land m+2q+3>0)\right) \bigvee \left(\frac{m+2q+1}{2} \in \mathbb{Z}^- \bigwedge \frac{m-1}{2} \notin \mathbb{Z}^-\right)$, **let** $u = \int (f x)^m \left(d+e x^2\right)^q dx$, **then**

$$\int (f x)^m \left(d+e x^2\right)^q (a+b \operatorname{ArcTan}[c x]) \, dx \longrightarrow u (a+b \operatorname{ArcTan}[c x]) - b c \int \frac{u}{1+c^2 x^2} \, dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^q,x]},
  Dist[a+b*ArcTan[c*x],u,x] - b*c*Int[SimplifyIntegrand[u/(1+c^2*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && (
  IGtQ[q,0] && Not[ILtQ[(m-1)/2,0] && GtQ[m+2*q+3,0]] ||
  IGtQ[(m+1)/2,0] && Not[ILtQ[q,0] && GtQ[m+2*q+3,0]] ||
  ILtQ[(m+2*q+1)/2,0] && Not[ILtQ[(m-1)/2,0]] )
```

```
Int[(f_.*x_)^m_.*(d_.+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[(f*x)^m*(d+e*x^2)^q,x]},
  Dist[a+b*ArcCot[c*x],u,x] + b*c*Int[SimplifyIntegrand[u/(1+c^2*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,m,q},x] && (
  IGtQ[q,0] && Not[ILtQ[(m-1)/2,0] && GtQ[m+2*q+3,0]] ||
  IGtQ[(m+1)/2,0] && Not[ILtQ[q,0] && GtQ[m+2*q+3,0]] ||
  ILtQ[(m+2*q+1)/2,0] && Not[ILtQ[(m-1)/2,0]] )
```

4: $\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{\left(d+e\,x^2\right)^2}\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{x}{(d+e\,x^2)^2} = \frac{1}{4\,d^2\sqrt{-\frac{e}{d}}\left(1-\sqrt{-\frac{e}{d}}\,x\right)^2} - \frac{1}{4\,d^2\sqrt{-\frac{e}{d}}\left(1+\sqrt{-\frac{e}{d}}\,x\right)^2}$

Rule: If $p\in\mathbb{Z}^+$, then

$$\int \frac{x\,(a+b\,\text{ArcTan}[c\,x])^p}{\left(d+e\,x^2\right)^2}\,dx \to \frac{1}{4\,d^2\sqrt{-\frac{e}{d}}}\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{\left(1-\sqrt{-\frac{e}{d}}\,x\right)^2}\,dx - \frac{1}{4\,d^2\sqrt{-\frac{e}{d}}}\int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{\left(1+\sqrt{-\frac{e}{d}}\,x\right)^2}\,dx$$

Program code:

```
Int[x_*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2)^2,x_Symbol] :=
  1/(4*d^2*Rt[-e/d,2])*Int[(a+b*ArcTan[c*x])^p/(1-Rt[-e/d,2]*x)^2,x] -
  1/(4*d^2*Rt[-e/d,2])*Int[(a+b*ArcTan[c*x])^p/(1+Rt[-e/d,2]*x)^2,x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0]
```

```
Int[x_*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2)^2,x_Symbol] :=
  1/(4*d^2*Rt[-e/d,2])*Int[(a+b*ArcCot[c*x])^p/(1-Rt[-e/d,2]*x)^2,x] -
  1/(4*d^2*Rt[-e/d,2])*Int[(a+b*ArcCot[c*x])^p/(1+Rt[-e/d,2]*x)^2,x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0]
```

5: $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx$ when $q\in\mathbb{Z}\,\land\,p\in\mathbb{Z}^+\land\,(p==1\,\lor\,m\in\mathbb{Z})$

Derivation: Algebraic expansion

Rule: If $q\in\mathbb{Z}\,\land\,p\in\mathbb{Z}^+\land\,(p==1\,\lor\,m\in\mathbb{Z})$**, then**

$$\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])^p\,dx\;\longrightarrow\;\int (a+b\,\text{ArcTan}[c\,x])^p\,\text{ExpandIntegrand}\left[(f\,x)^m \left(d+e\,x^2\right)^q,\,x\right]dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*ArcTan[c*x])^p,(f*x)^m*(d+e*x^2)^q,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,m},x] && IntegerQ[q] && IGtQ[p,0] && (EqQ[p,1] && GtQ[q,0] || IntegerQ[m])
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*ArcCot[c*x])^p,(f*x)^m*(d+e*x^2)^q,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,m},x] && IntegerQ[q] && IGtQ[p,0] && (EqQ[p,1] && GtQ[q,0] || IntegerQ[m])
```

6: $\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])\,dx$

Derivation: Algebraic expansion

Rule:

$$\int (f\,x)^m \left(d+e\,x^2\right)^q (a+b\,\text{ArcTan}[c\,x])\,dx\;\longrightarrow\;a\int (f\,x)^m \left(d+e\,x^2\right)^q dx+b\int (f\,x)^m \left(d+e\,x^2\right)^q \text{ArcTan}[c\,x]\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  a*Int[(f*x)^m*(d+e*x^2)^q,x] + b*Int[(f*x)^m*(d+e*x^2)^q*ArcTan[c*x],x] /;
FreeQ[{a,b,c,d,e,f,m,q},x]
```

```
Int[(f_.*x_)^m_.*(d_+e_.*x_^2)^q_.*(a_+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  a*Int[(f*x)^m*(d+e*x^2)^q,x] + b*Int[(f*x)^m*(d+e*x^2)^q*ArcCot[c*x],x] /;
FreeQ[{a,b,c,d,e,f,m,q},x]
```

7. $\int \frac{u\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $e == c^2\,d$

1: $\int \frac{(f+g\,x)^m\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $p \in \mathbb{Z}^+ \wedge e == c^2\,d \wedge m \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+ \wedge e == c^2\,d \wedge m \in \mathbb{Z}^+$**, then**

$$\int \frac{(f+g\,x)^m\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx \longrightarrow \int \frac{(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,\text{ExpandIntegrand}[(f+g\,x)^m,\,x]\,dx$$

Program code:

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcTan[c*x])^p/(d+e*x^2),(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[p,0] && EqQ[e,c^2*d] && IGtQ[m,0]
```

```
Int[(f_+g_.*x_)^m_.*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[(a+b*ArcCot[c*x])^p/(d+e*x^2),(f+g*x)^m,x],x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[p,0] && EqQ[e,c^2*d] && IGtQ[m,0]
```

2. $\int \frac{\text{ArcTanh}[u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $p \in \mathbb{Z}^+ \bigwedge e == c^2 d$

1: $\int \frac{\text{ArcTanh}[u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $p \in \mathbb{Z}^+ \bigwedge e == c^2 d \bigwedge u^2 == \left(1-\frac{2\,I}{I+c\,x}\right)^2$

Derivation: Algebraic expansion

- **Basis:** $\text{ArcTanh}[z] == \frac{1}{2}\,\text{Log}[1+z] - \frac{1}{2}\,\text{Log}[1-z]$

- **Basis:** $\text{ArcCoth}[z] == \frac{1}{2}\,\text{Log}\left[1+\frac{1}{z}\right] - \frac{1}{2}\,\text{Log}\left[1-\frac{1}{z}\right]$

- **Rule: If** $p \in \mathbb{Z}^+ \bigwedge e == c^2 d \bigwedge u^2 == \left(1-\frac{2\,I}{I+c\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTanh}[u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx \longrightarrow \frac{1}{2}\int \frac{\text{Log}[1+u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx - \frac{1}{2}\int \frac{\text{Log}[1-u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$$

Program code:

```
Int[ArcTanh[u_]*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  1/2*Int[Log[1+u]*(a+b*ArcTan[c*x])^p/(d+e*x^2),x] -
  1/2*Int[Log[1-u]*(a+b*ArcTan[c*x])^p/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[u^2-(1-2*I/(I+c*x))^2,0]
```

```
Int[ArcCoth[u_]*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  1/2*Int[Log[SimplifyIntegrand[1+1/u,x]]*(a+b*ArcCot[c*x])^p/(d+e*x^2),x] -
  1/2*Int[Log[SimplifyIntegrand[1-1/u,x]]*(a+b*ArcCot[c*x])^p/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[u^2-(1-2*I/(I+c*x))^2,0]
```

2: $\int \frac{\text{ArcTanh}[u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $p \in \mathbb{Z}^+ \land e == c^2 d \land u^2 == \left(1-\frac{2\,I}{I-c\,x}\right)^2$

Derivation: Algebraic expansion

Basis: $\text{ArcTanh}[z] == \frac{1}{2}\,\text{Log}[1+z] - \frac{1}{2}\,\text{Log}[1-z]$

Basis: $\text{ArcCoth}[z] == \frac{1}{2}\,\text{Log}\left[1+\frac{1}{z}\right] - \frac{1}{2}\,\text{Log}\left[1-\frac{1}{z}\right]$

Rule: If $p \in \mathbb{Z}^+ \land e == c^2 d \land u^2 == \left(1-\frac{2\,I}{I-c\,x}\right)^2$**, then**

$$\int \frac{\text{ArcTanh}[u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx \longrightarrow \frac{1}{2}\int \frac{\text{Log}[1+u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx - \frac{1}{2}\int \frac{\text{Log}[1-u]\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$$

Program code:

```
Int[ArcTanh[u_]*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  1/2*Int[Log[1+u]*(a+b*ArcTan[c*x])^p/(d+e*x^2),x] -
  1/2*Int[Log[1-u]*(a+b*ArcTan[c*x])^p/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[u^2-(1-2*I/(I-c*x))^2,0]
```

```
Int[ArcCoth[u_]*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  1/2*Int[Log[SimplifyIntegrand[1+1/u,x]]*(a+b*ArcCot[c*x])^p/(d+e*x^2),x] -
  1/2*Int[Log[SimplifyIntegrand[1-1/u,x]]*(a+b*ArcCot[c*x])^p/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[u^2-(1-2*I/(I-c*x))^2,0]
```

3. $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{Log}[u]}{d+e\,x^2}\,dx$ when $p\in\mathbb{Z}^+\wedge e==c^2\,d$

1: $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{Log}[f+g\,x]}{d+e\,x^2}\,dx$ when $p\in\mathbb{Z}^+\wedge e==c^2\,d\wedge c^2\,f^2+g^2==0$

Derivation: Integration by parts

Basis: If $e==c^2\,d$**, then** $\frac{(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}==\partial_x\frac{(a+b\,\text{ArcTan}[c\,x])^{p+1}}{b\,c\,d\,(p+1)}$

Rule: If $p\in\mathbb{Z}^+\wedge e==c^2\,d\wedge c^2\,f^2+g^2==0$**, then**

$$\int\frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{Log}[f+g\,x]}{d+e\,x^2}\,dx\longrightarrow\frac{(a+b\,\text{ArcTan}[c\,x])^{p+1}\,\text{Log}[f+g\,x]}{b\,c\,d\,(p+1)}-\frac{g}{b\,c\,d\,(p+1)}\int\frac{(a+b\,\text{ArcTan}[c\,x])^{p+1}}{f+g\,x}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_.*Log[f_+g_.*x_]/(d_+e_.*x_^2),x_Symbol] :=
  (a+b*ArcTan[c*x])^(p+1)*Log[f+g*x]/(b*c*d*(p+1)) -
  g/(b*c*d*(p+1))*Int[(a+b*ArcTan[c*x])^(p+1)/(f+g*x),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[c^2*f^2+g^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_.*Log[f_+g_.*x_]/(d_+e_.*x_^2),x_Symbol] :=
  (a+b*ArcCot[c*x])^(p+1)*Log[f+g*x]/(b*c*d*(p+1)) -
  g/(b*c*d*(p+1))*Int[(a+b*ArcCot[c*x])^(p+1)/(f+g*x),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[c^2*f^2+g^2,0]
```

2: $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{Log}[u]}{d+e\,x^2}\,dx$ when $p\in\mathbb{Z}^+\bigwedge e==c^2\,d\bigwedge (1-u)^2==\left(1-\frac{2\,I}{I+c\,x}\right)^2$

Derivation: Integration by parts

- **Rule: If** $p\in\mathbb{Z}^+\bigwedge e==c^2\,d\bigwedge (1-u)^2==\left(1-\frac{2\,I}{I+c\,x}\right)^2$**, then**

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{Log}[u]}{d+e\,x^2}\,dx\to\frac{i\,(a+b\,\text{ArcTan}[c\,x])^p\,\text{PolyLog}[2,\,1-u]}{2\,c\,d}-\frac{b\,p\,i}{2}\int\frac{(a+b\,\text{ArcTan}[c\,x])^{p-1}\,\text{PolyLog}[2,\,1-u]}{d+e\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_.*Log[u_]/(d_+e_.*x_^2),x_Symbol] :=
  I*(a+b*ArcTan[c*x])^p*PolyLog[2,1-u]/(2*c*d) -
  b*p*I/2*Int[(a+b*ArcTan[c*x])^(p-1)*PolyLog[2,1-u]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[(1-u)^2-(1-2*I/(I+c*x))^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_.*Log[u_]/(d_+e_.*x_^2),x_Symbol] :=
  I*(a+b*ArcCot[c*x])^p*PolyLog[2,1-u]/(2*c*d) +
  b*p*I/2*Int[(a+b*ArcCot[c*x])^(p-1)*PolyLog[2,1-u]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[(1-u)^2-(1-2*I/(I+c*x))^2,0]
```

3: $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{Log}[u]}{d+e\,x^2}\,dx$ when $p\in\mathbb{Z}^+\bigwedge e==c^2\,d\bigwedge (1-u)^2==\left(1-\frac{2\,I}{I-c\,x}\right)^2$

Derivation: Integration by parts

- **Rule: If** $p\in\mathbb{Z}^+\bigwedge e==c^2\,d\bigwedge (1-u)^2==\left(1-\frac{2\,I}{I-c\,x}\right)^2$**, then**

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{Log}[u]}{d+e\,x^2}\,dx\to-\frac{i\,(a+b\,\text{ArcTan}[c\,x])^p\,\text{PolyLog}[2,\,1-u]}{2\,c\,d}+\frac{b\,p\,i}{2}\int\frac{(a+b\,\text{ArcTan}[c\,x])^{p-1}\,\text{PolyLog}[2,\,1-u]}{d+e\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_.*Log[u_]/(d_+e_.*x_^2),x_Symbol] :=
  -I*(a+b*ArcTan[c*x])^p*PolyLog[2,1-u]/(2*c*d) +
  b*p*I/2*Int[(a+b*ArcTan[c*x])^(p-1)*PolyLog[2,1-u]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[(1-u)^2-(1-2*I/(I-c*x))^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_.*Log[u_]/(d_+e_.*x_^2),x_Symbol] :=
  -I*(a+b*ArcCot[c*x])^p*PolyLog[2,1-u]/(2*c*d) -
  b*p*I/2*Int[(a+b*ArcCot[c*x])^(p-1)*PolyLog[2,1-u]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[(1-u)^2-(1-2*I/(I-c*x))^2,0]
```

4. $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{PolyLog}[k,u]}{d+e\,x^2}\,dx$ when $p \in \mathbb{Z}^+ \wedge e == c^2\,d$

1: $\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{PolyLog}[k,u]}{d+e\,x^2}\,dx$ when $p \in \mathbb{Z}^+ \bigwedge e == c^2\,d \bigwedge u^2 == \left(1-\frac{2\,I}{I+c\,x}\right)^2$

Derivation: Integration by parts

- **Rule: If** $p \in \mathbb{Z}^+ \bigwedge e == c^2\,d \bigwedge u^2 == \left(1-\frac{2\,I}{I+c\,x}\right)^2$**, then**

$$\int \frac{(a+b\,\text{ArcTan}[c\,x])^p\,\text{PolyLog}[k,u]}{d+e\,x^2}\,dx \longrightarrow -\frac{i\,(a+b\,\text{ArcTan}[c\,x])^p\,\text{PolyLog}[k+1,u]}{2\,c\,d}+\frac{b\,p\,i}{2}\int \frac{(a+b\,\text{ArcTan}[c\,x])^{p-1}\,\text{PolyLog}[k+1,u]}{d+e\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_.*PolyLog[k_,u_]/(d_+e_.*x_^2),x_Symbol] :=
  -I*(a+b*ArcTan[c*x])^p*PolyLog[k+1,u]/(2*c*d) +
  b*p*I/2*Int[(a+b*ArcTan[c*x])^(p-1)*PolyLog[k+1,u]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,k},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[u^2-(1-2*I/(I+c*x))^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_.*PolyLog[k_,u_]/(d_+e_.*x_^2),x_Symbol] :=
  -I*(a+b*ArcCot[c*x])^p*PolyLog[k+1,u]/(2*c*d) -
  b*p*I/2*Int[(a+b*ArcCot[c*x])^(p-1)*PolyLog[k+1,u]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,k},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[u^2-(1-2*I/(I+c*x))^2,0]
```

2: $\int \frac{(a + b \operatorname{ArcTan}[c\,x])^p \operatorname{PolyLog}[k,\,u]}{d + e\,x^2}\,dx$ when $p \in \mathbb{Z}^+ \bigwedge e == c^2 d \bigwedge u^2 == \left(1 - \frac{2\,I}{I - c\,x}\right)^2$

Derivation: Integration by parts

Rule: If $p \in \mathbb{Z}^+ \bigwedge e == c^2 d \bigwedge u^2 == \left(1 - \frac{2\,I}{I - c\,x}\right)^2$, then

$$\int \frac{(a + b \operatorname{ArcTan}[c\,x])^p \operatorname{PolyLog}[k,\,u]}{d + e\,x^2}\,dx \longrightarrow \frac{i\,(a + b \operatorname{ArcTan}[c\,x])^p \operatorname{PolyLog}[k + 1,\,u]}{2\,c\,d} - \frac{b\,p\,i}{2} \int \frac{(a + b \operatorname{ArcTan}[c\,x])^{p-1} \operatorname{PolyLog}[k + 1,\,u]}{d + e\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcTan[c_.*x_])^p_.*PolyLog[k_,u_]/(d_+e_.*x_^2),x_Symbol] :=
  I*(a+b*ArcTan[c*x])^p*PolyLog[k+1,u]/(2*c*d) -
  b*p*I/2*Int[(a+b*ArcTan[c*x])^(p-1)*PolyLog[k+1,u]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,k},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[u^2-(1-2*I/(I-c*x))^2,0]
```

```
Int[(a_.+b_.*ArcCot[c_.*x_])^p_.*PolyLog[k_,u_]/(d_+e_.*x_^2),x_Symbol] :=
  I*(a+b*ArcCot[c*x])^p*PolyLog[k+1,u]/(2*c*d) +
  b*p*I/2*Int[(a+b*ArcCot[c*x])^(p-1)*PolyLog[k+1,u]/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e,k},x] && IGtQ[p,0] && EqQ[e,c^2*d] && EqQ[u^2-(1-2*I/(I-c*x))^2,0]
```

5. $\int \frac{(a + b \operatorname{ArcCot}[c\,x])^q\,(a + b \operatorname{ArcTan}[c\,x])^p}{d + e\,x^2}\,dx$ when $e == c^2 d$

1: $\int \frac{1}{(d + e\,x^2)\,(a + b \operatorname{ArcCot}[c\,x])\,(a + b \operatorname{ArcTan}[c\,x])}\,dx$ when $e == c^2 d$

Rule: If $e == c^2 d$, then

$$\int \frac{1}{(d + e\,x^2)\,(a + b \operatorname{ArcCot}[c\,x])\,(a + b \operatorname{ArcTan}[c\,x])}\,dx \longrightarrow \frac{-\operatorname{Log}[a + b \operatorname{ArcCot}[c\,x]] + \operatorname{Log}[a + b \operatorname{ArcTan}[c\,x]]}{b\,c\,d\,(2\,a + b \operatorname{ArcCot}[c\,x] + b \operatorname{ArcTan}[c\,x])}$$

Program code:

```
Int[1/((d_+e_.*x_^2)*(a_.+b_.*ArcCot[c_.*x_])*(a_.+b_.*ArcTan[c_.*x_])),x_Symbol] :=
  (-Log[a+b*ArcCot[c*x]]+Log[a+b*ArcTan[c*x]])/(b*c*d*(2*a+b*ArcCot[c*x]+b*ArcTan[c*x])) /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d]
```

2: $\int \frac{(a+b\,\text{ArcCot}[c\,x])^q\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx$ when $e == c^2 d\ \wedge\ (p \mid q) \in \mathbb{Z}\ \wedge\ 0 < p \le q$

Derivation: Integration by parts

Rule: If $e == c^2 d\ \wedge\ (p \mid q) \in \mathbb{Z}\ \wedge\ 0 < p \le q$**, then**

$$\int \frac{(a+b\,\text{ArcCot}[c\,x])^q\,(a+b\,\text{ArcTan}[c\,x])^p}{d+e\,x^2}\,dx \longrightarrow -\frac{(a+b\,\text{ArcCot}[c\,x])^{q+1}\,(a+b\,\text{ArcTan}[c\,x])^p}{b\,c\,d\,(q+1)} + \frac{p}{q+1}\int \frac{(a+b\,\text{ArcCot}[c\,x])^{q+1}\,(a+b\,\text{ArcTan}[c\,x])^{p-1}}{d+e\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*ArcCot[c_.*x_])^q_.*(a_.+b_.*ArcTan[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  -(a+b*ArcCot[c*x])^(q+1)*(a+b*ArcTan[c*x])^p/(b*c*d*(q+1)) +
  p/(q+1)*Int[(a+b*ArcCot[c*x])^(q+1)*(a+b*ArcTan[c*x])^(p-1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && IGeQ[q,p]
```

```
Int[(a_.+b_.*ArcTan[c_.*x_])^q_.*(a_.+b_.*ArcCot[c_.*x_])^p_./(d_+e_.*x_^2),x_Symbol] :=
  (a+b*ArcTan[c*x])^(q+1)*(a+b*ArcCot[c*x])^p/(b*c*d*(q+1)) +
  p/(q+1)*Int[(a+b*ArcTan[c*x])^(q+1)*(a+b*ArcCot[c*x])^(p-1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[e,c^2*d] && IGtQ[p,0] && IGeQ[q,p]
```

8: $\int \frac{\text{ArcTan}[a\,x]}{c+d\,x^n}\,dx$ when $n \in \mathbb{Z}\ \wedge\ \neg\left(n == 2 \wedge d == a^2 c\right)$

Derivation: Algebraic expansion

Basis: $\text{ArcTan}[z] == \frac{1}{2}\,i\,\text{Log}[1-i\,z] - \frac{1}{2}\,i\,\text{Log}[1+i\,z]$

Basis: $\text{ArcCot}[z] == \frac{1}{2}\,i\,\text{Log}\left[1-\frac{i}{z}\right] - \frac{1}{2}\,i\,\text{Log}\left[1+\frac{i}{z}\right]$

Rule: If $n \in \mathbb{Z}\ \wedge\ \neg\left(n == 2 \wedge d == a^2 c\right)$**, then**

$$\int \frac{\text{ArcTan}[a\,x]}{c+d\,x^n}\,dx \longrightarrow \frac{i}{2}\int \frac{\text{Log}[1-i\,a\,x]}{c+d\,x^n}\,dx - \frac{i}{2}\int \frac{\text{Log}[1+i\,a\,x]}{c+d\,x^n}\,dx$$

Program code:

```
Int[ArcTan[a_.*x_]/(c_+d_.*x_^n_.),x_Symbol] :=
  I/2*Int[Log[1-I*a*x]/(c+d*x^n),x] -
  I/2*Int[Log[1+I*a*x]/(c+d*x^n),x] /;
FreeQ[{a,c,d},x] && IntegerQ[n] && Not[EqQ[n,2] && EqQ[d,a^2*c]]
```

```
Int[ArcCot[a_.*x_]/(c_+d_.*x_^n_.),x_Symbol] :=
  I/2*Int[Log[1-I/(a*x)]/(c+d*x^n),x] -
  I/2*Int[Log[1+I/(a*x)]/(c+d*x^n),x] /;
FreeQ[{a,c,d},x] && IntegerQ[n] && Not[EqQ[n,2]] && EqQ[d,a^2*c]]
```

9. $\int \frac{\text{Log}[d\,x^m]\,(a+b\,\text{ArcTan}[c\,x^n])}{x}\,dx$

1: $\int \frac{\text{Log}[d\,x^m]\,\text{ArcTan}[c\,x^n]}{x}\,dx$

Derivation: Algebraic expansion

Basis: $\text{ArcTan}[c\,x^n] = \frac{i}{2}\,\text{Log}[1-i\,c\,x^n] - \frac{i}{2}\,\text{Log}[1+i\,c\,x^n]$

Rule:

$$\int \frac{\text{Log}[d\,x^m]\,\text{ArcTan}[c\,x^n]}{x}\,dx \longrightarrow \frac{i}{2}\int \frac{\text{Log}[d\,x^m]\,\text{Log}[1-i\,c\,x^n]}{x}\,dx - \frac{i}{2}\int \frac{\text{Log}[d\,x^m]\,\text{Log}[1+i\,c\,x^n]}{x}\,dx$$

Program code:

```
Int[Log[d_.*x_^m_.]*ArcTan[c_.*x_^n_.]/x_,x_Symbol] :=
  I/2*Int[Log[d*x^m]*Log[1-I*c*x^n]/x,x] - I/2*Int[Log[d*x^m]*Log[1+I*c*x^n]/x,x] /;
FreeQ[{c,d,m,n},x]
```

```
Int[Log[d_.*x_^m_.]*ArcCot[c_.*x_^n_.]/x_,x_Symbol] :=
  I/2*Int[Log[d*x^m]*Log[1-I/(c*x^n)]/x,x] - I/2*Int[Log[d*x^m]*Log[1+I/(c*x^n)]/x,x] /;
FreeQ[{c,d,m,n},x]
```

2: $\int \frac{\text{Log}[d\, x^m]\, (a + b\, \text{ArcTan}[c\, x^n])}{x}\, dx$

Derivation: Algebraic expansion

Rule:

$$\int \frac{\text{Log}[d\, x^m]\, (a + b\, \text{ArcTan}[c\, x^n])}{x}\, dx \longrightarrow a \int \frac{\text{Log}[d\, x^m]}{x}\, dx + b \int \frac{\text{Log}[d\, x^m]\, \text{ArcTan}[c\, x^n]}{x}\, dx$$

Program code:

```
Int[Log[d_.*x_^m_.]*(a_+b_.*ArcTan[c_.*x_^n_.])/x_,x_Symbol] :=
  a*Int[Log[d*x^m]/x,x] + b*Int[(Log[d*x^m]*ArcTan[c*x^n])/x,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

```
Int[Log[d_.*x_^m_.]*(a_+b_.*ArcCot[c_.*x_^n_.])/x_,x_Symbol] :=
  a*Int[Log[d*x^m]/x,x] + b*Int[(Log[d*x^m]*ArcCot[c*x^n])/x,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

10. $\int u\, (d + e\, \text{Log}[f + g\, x^2])\, (a + b\, \text{ArcTan}[c\, x])^p\, dx$

1: $\int (d + e\, \text{Log}[f + g\, x^2])\, (a + b\, \text{ArcTan}[c\, x])\, dx$

Derivation: Integration by parts

Rule:

$$\int (d + e\, \text{Log}[f + g\, x^2])\, (a + b\, \text{ArcTan}[c\, x])\, dx \longrightarrow$$

$$x\, (d + e\, \text{Log}[f + g\, x^2])\, (a + b\, \text{ArcTan}[c\, x]) - 2\, e\, g \int \frac{x^2\, (a + b\, \text{ArcTan}[c\, x])}{f + g\, x^2}\, dx - b\, c \int \frac{x\, (d + e\, \text{Log}[f + g\, x^2])}{1 + c^2\, x^2}\, dx$$

Program code:

```
Int[(d_.+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  x*(d+e*Log[f+g*x^2])*(a+b*ArcTan[c*x]) -
  2*e*g*Int[x^2*(a+b*ArcTan[c*x])/(f+g*x^2),x] -
  b*c*Int[x*(d+e*Log[f+g*x^2])/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x]
```

```
Int[(d_.+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  x*(d+e*Log[f+g*x^2])*(a+b*ArcCot[c*x]) -
  2*e*g*Int[x^2*(a+b*ArcCot[c*x])/(f+g*x^2),x] +
  b*c*Int[x*(d+e*Log[f+g*x^2])/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x]
```

2. $\int x^m \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x]) \, dx$

1. $\int \frac{\left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x])}{x} \, dx$

1. $\int \frac{\operatorname{Log}\left[f + g x^2\right] (a + b \operatorname{ArcTan}[c x])}{x} \, dx$

1. $\int \frac{\operatorname{Log}\left[f + g x^2\right] \operatorname{ArcTan}[c x]}{x} \, dx$ when $c^2 f + g == 0$

1: $\int \frac{\operatorname{Log}\left[f + g x^2\right] \operatorname{ArcTan}[c x]}{x} \, dx$ when $g == c^2 f$

Derivation: Piecewise constant extraction and algebraic simplification

Basis: If $g == c^2 f$, **then** $\partial_x \left(\operatorname{Log}\left[f + g x^2\right] - \operatorname{Log}[1 - i c x] - \operatorname{Log}[1 + i c x]\right) == 0$

Basis: $(\operatorname{Log}[1 - i c x] + \operatorname{Log}[1 + i c x]) \operatorname{ArcTan}[c x] == \frac{i}{2} \operatorname{Log}[1 - i c x]^2 - \frac{i}{2} \operatorname{Log}[1 + i c x]^2$

Rule: If $g == c^2 f$, **then**

$$\int \frac{\operatorname{Log}\left[f + g x^2\right] \operatorname{ArcTan}[c x]}{x} \, dx \longrightarrow$$

$$\left(\operatorname{Log}\left[f + g x^2\right] - \operatorname{Log}[1 - i c x] - \operatorname{Log}[1 + i c x]\right) \int \frac{\operatorname{ArcTan}[c x]}{x} \, dx + \int \frac{(\operatorname{Log}[1 - i c x] + \operatorname{Log}[1 + i c x]) \operatorname{ArcTan}[c x]}{x} \, dx \longrightarrow$$

$$\left(\operatorname{Log}\left[f + g x^2\right] - \operatorname{Log}[1 - i c x] - \operatorname{Log}[1 + i c x]\right) \int \frac{\operatorname{ArcTan}[c x]}{x} \, dx + \frac{i}{2} \int \frac{\operatorname{Log}[1 - i c x]^2}{x} \, dx - \frac{i}{2} \int \frac{\operatorname{Log}[1 + i c x]^2}{x} \, dx$$

```
Int[Log[f_.+g_.*x_^2]*ArcTan[c_.*x_]/x_,x_Symbol] :=
  (Log[f+g*x^2]-Log[1-I*c*x]-Log[1+I*c*x])*Int[ArcTan[c*x]/x,x] + I/2*Int[Log[1-I*c*x]^2/x,x] - I/2*Int[Log[1+I*c*x]^2/x,x] /;
FreeQ[{c,f,g},x] && EqQ[g,c^2*f]
```

2: $\int \frac{\text{Log}[f + g x^2] \text{ ArcCot}[c x]}{x} \, dx$ when $g == c^2 f$

Derivation: Piecewise constant extraction and algebraic simplification

- **Basis: If** $g == c^2 f$**, then** $\partial_x \left(\text{Log}[f + g x^2] - \text{Log}[c^2 x^2] - \text{Log}\left[1 - \frac{i}{c x}\right] - \text{Log}\left[1 + \frac{i}{c x}\right]\right) == 0$
- **Basis:** $\left(\text{Log}[c^2 x^2] + \text{Log}\left[1 - \frac{i}{c x}\right] + \text{Log}\left[1 + \frac{i}{c x}\right]\right) \text{ArcCot}[c x] == \text{Log}[c^2 x^2] \text{ ArcCot}[c x] + \frac{i}{2} \text{Log}\left[1 - \frac{i}{c x}\right]^2 - \frac{i}{2} \text{Log}\left[1 + \frac{i}{c x}\right]^2$
- **Rule: If** $g == c^2 f$**, then**

$$\int \frac{\text{Log}[f + g x^2] \text{ ArcCot}[c x]}{x} \, dx \longrightarrow$$

$$\left(\text{Log}[f + g x^2] - \text{Log}[c^2 x^2] - \text{Log}\left[1 - \frac{i}{c x}\right] - \text{Log}\left[1 + \frac{i}{c x}\right]\right) \int \frac{\text{ArcCot}[c x]}{x} \, dx + \int \frac{\left(\text{Log}[c^2 x^2] + \text{Log}\left[1 - \frac{i}{c x}\right] + \text{Log}\left[1 + \frac{i}{c x}\right]\right) \text{ArcCot}[c x]}{x} \, dx \longrightarrow$$

$$\left(\text{Log}[f + g x^2] - \text{Log}[c^2 x^2] - \text{Log}\left[1 - \frac{i}{c x}\right] - \text{Log}\left[1 + \frac{i}{c x}\right]\right) \int \frac{\text{ArcCot}[c x]}{x} \, dx + \int \frac{\text{Log}[c^2 x^2] \text{ ArcCot}[c x]}{x} \, dx + \frac{i}{2} \int \frac{\text{Log}\left[1 - \frac{i}{c x}\right]^2}{x} \, dx - \frac{i}{2} \int \frac{\text{Log}\left[1 + \frac{i}{c x}\right]^2}{x} \, dx$$

Program code:

```
Int[Log[f_.+g_.*x_^2]*ArcCot[c_.*x_]/x_,x_Symbol] :=
  (Log[f+g*x^2]-Log[c^2*x^2]-Log[1-I/(c*x)]-Log[1+I/(c*x)])*Int[ArcCot[c*x]/x,x] +
  Int[Log[c^2*x^2]*ArcCot[c*x]/x,x] +
  I/2*Int[Log[1-I/(c*x)]^2/x,x] -
  I/2*Int[Log[1+I/(c*x)]^2/x,x] /;
FreeQ[{c,f,g},x] && EqQ[g,c^2*f]
```

2: $\int \frac{\text{Log}[f+g x^2]\,(a+b\,\text{ArcTan}[c x])}{x}\,dx$

Derivation: Algebraic expansion

Rule:

$$\int \frac{\text{Log}[f+g x^2]\,(a+b\,\text{ArcTan}[c x])}{x}\,dx \longrightarrow a\int \frac{\text{Log}[f+g x^2]}{x}\,dx+b\int \frac{\text{Log}[f+g x^2]\,\text{ArcTan}[c x]}{x}\,dx$$

Program code:

```
Int[Log[f_.+g_.*x_^2]*(a_.+b_.*ArcTan[c_.*x_])/x_,x_Symbol] :=
  a*Int[Log[f+g*x^2]/x,x] + b*Int[Log[f+g*x^2]*ArcTan[c*x]/x,x] /;
FreeQ[{a,b,c,f,g},x]
```

```
Int[Log[f_.+g_.*x_^2]*(a_.+b_.*ArcCot[c_.*x_])/x_,x_Symbol] :=
  a*Int[Log[f+g*x^2]/x,x] + b*Int[Log[f+g*x^2]*ArcCot[c*x]/x,x] /;
FreeQ[{a,b,c,f,g},x]
```

2: $\int \frac{(d+e\,\text{Log}[f+g x^2])\,(a+b\,\text{ArcTan}[c x])}{x}\,dx$

Derivation: Algebraic expansion

Rule:

$$\int \frac{(d+e\,\text{Log}[f+g x^2])\,(a+b\,\text{ArcTan}[c x])}{x}\,dx \longrightarrow d\int \frac{a+b\,\text{ArcTan}[c x]}{x}\,dx+e\int \frac{\text{Log}[f+g x^2]\,(a+b\,\text{ArcTan}[c x])}{x}\,dx$$

Program code:

```
Int[(d_+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcTan[c_.*x_])/x_,x_Symbol] :=
  d*Int[(a+b*ArcTan[c*x])/x,x] + e*Int[Log[f+g*x^2]*(a+b*ArcTan[c*x])/x,x] /;
FreeQ[{a,b,c,d,e,f,g},x]
```

```
Int[(d_+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcCot[c_.*x_])/x_,x_Symbol] :=
  d*Int[(a+b*ArcCot[c*x])/x,x] + e*Int[Log[f+g*x^2]*(a+b*ArcCot[c*x])/x,x] /;
FreeQ[{a,b,c,d,e,f,g},x]
```

2: $\int x^m \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x]) \, dx$ when $\frac{m}{2} \in \mathbb{Z}^-$

Derivation: Integration by parts

Rule: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\int x^m \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x]) \, dx \longrightarrow$$
$$\frac{x^{m+1} \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x])}{m+1} - \frac{2 e g}{m+1} \int \frac{x^{m+2} (a + b \operatorname{ArcTan}[c x])}{f + g x^2} \, dx - \frac{b c}{m+1} \int \frac{x^{m+1} \left(d + e \operatorname{Log}\left[f + g x^2\right]\right)}{1 + c^2 x^2} \, dx$$

Program code:

```
Int[x_^m_.*(d_.+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  x^(m+1)*(d+e*Log[f+g*x^2])*(a+b*ArcTan[c*x])/(m+1) -
  2*e*g/(m+1)*Int[x^(m+2)*(a+b*ArcTan[c*x])/(f+g*x^2),x] -
  b*c/(m+1)*Int[x^(m+1)*(d+e*Log[f+g*x^2])/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && ILtQ[m/2,0]
```

```
Int[x_^m_.*(d_.+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  x^(m+1)*(d+e*Log[f+g*x^2])*(a+b*ArcCot[c*x])/(m+1) -
  2*e*g/(m+1)*Int[x^(m+2)*(a+b*ArcCot[c*x])/(f+g*x^2),x] +
  b*c/(m+1)*Int[x^(m+1)*(d+e*Log[f+g*x^2])/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && ILtQ[m/2,0]
```

3: $\int x^m \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x]) \, dx$ when $\frac{m+1}{2} \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If $\frac{m+1}{2} \in \mathbb{Z}^+$, let $u = \int x^m \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) dx$, then

$$\int x^m \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x]) \, dx \longrightarrow u \, (a + b \operatorname{ArcTan}[c x]) - b c \int \frac{u}{1 + c^2 x^2} \, dx$$

Program code:

```
Int[x_^m_.*(d_.+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*Log[f+g*x^2]),x]},
  Dist[a+b*ArcTan[c*x],u,x] - b*c*Int[ExpandIntegrand[u/(1+c^2*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[(m+1)/2,0]
```

```
Int[x_^m_.*(d_.+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(d+e*Log[f+g*x^2]),x]},
  Dist[a+b*ArcCot[c*x],u,x] + b*c*Int[ExpandIntegrand[u/(1+c^2*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IGtQ[(m+1)/2,0]
```

4: $\int x^m \left(d+e\,\text{Log}\left[f+g\,x^2\right]\right)(a+b\,\text{ArcTan}[c\,x])\,dx$ when $m\in\mathbb{Z}$

Derivation: Integration by parts

Rule: If $m\in\mathbb{Z}$**, let** $u=\int x^m\,(a+b\,\text{ArcTan}[c\,x])\,dx$**, then**

$$\int x^m \left(d+e\,\text{Log}\left[f+g\,x^2\right]\right)(a+b\,\text{ArcTan}[c\,x])\,dx \longrightarrow u\left(d+e\,\text{Log}\left[f+g\,x^2\right]\right)-2\,e\,g\int\frac{x\,u}{f+g\,x^2}\,dx$$

Program code:

```
Int[x_^m_.*(d_.+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcTan[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(a+b*ArcTan[c*x]),x]},
  Dist[d+e*Log[f+g*x^2],u,x] - 2*e*g*Int[ExpandIntegrand[x*u/(f+g*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IntegerQ[m] && NeQ[m,-1]
```

```
Int[x_^m_.*(d_.+e_.*Log[f_.+g_.*x_^2])*(a_.+b_.*ArcCot[c_.*x_]),x_Symbol] :=
  With[{u=IntHide[x^m*(a+b*ArcCot[c*x]),x]},
  Dist[d+e*Log[f+g*x^2],u,x] - 2*e*g*Int[ExpandIntegrand[x*u/(f+g*x^2),x],x]] /;
FreeQ[{a,b,c,d,e,f,g},x] && IntegerQ[m] && NeQ[m,-1]
```

3: $\int x \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x])^2 \, dx$ when $g == c^2 f$

Derivation: Integration by parts

- **Basis:** $x \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) == \partial_x \left(\frac{(f+g x^2) (d+e \operatorname{Log}[f+g x^2])}{2 g} - \frac{e x^2}{2}\right)$

Rule: If $g == c^2 f$**, then**

$$\int x \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x])^2 \, dx \longrightarrow$$

$$\frac{\left(f + g x^2\right) \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x])^2}{2 g} - \frac{e x^2 (a + b \operatorname{ArcTan}[c x])^2}{2} -$$

$$\frac{b}{c} \int \left(d + e \operatorname{Log}\left[f + g x^2\right]\right) (a + b \operatorname{ArcTan}[c x]) \, dx + b c e \int \frac{x^2 (a + b \operatorname{ArcTan}[c x])}{1 + c^2 x^2} \, dx$$

Program code:

```
Int[x_*(d_.+e_.*Log[f_+g_.*x_^2])*(a_.+b_.*ArcTan[c_.*x_])^2,x_Symbol] :=
  (f+g*x^2)*(d+e*Log[f+g*x^2])*(a+b*ArcTan[c*x])^2/(2*g) -
  e*x^2*(a+b*ArcTan[c*x])^2/2 -
  b/c*Int[(d+e*Log[f+g*x^2])*(a+b*ArcTan[c*x]),x] +
  b*c*e*Int[x^2*(a+b*ArcTan[c*x])/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[g,c^2*f]
```

```
Int[x_*(d_.+e_.*Log[f_+g_.*x_^2])*(a_.+b_.*ArcCot[c_.*x_])^2,x_Symbol] :=
  (f+g*x^2)*(d+e*Log[f+g*x^2])*(a+b*ArcCot[c*x])^2/(2*g) -
  e*x^2*(a+b*ArcCot[c*x])^2/2 +
  b/c*Int[(d+e*Log[f+g*x^2])*(a+b*ArcCot[c*x]),x] -
  b*c*e*Int[x^2*(a+b*ArcCot[c*x])/(1+c^2*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g},x] && EqQ[g,c^2*f]
```

U: $\int u\,(a+b\,\text{ArcTan}[c\,x])^p\,dx$

- **Rule:**

$$\int u\,(a+b\,\text{ArcTan}[c\,x])^p\,dx\ \longrightarrow\ \int u\,(a+b\,\text{ArcTan}[c\,x])^p\,dx$$

- **Program code:**

```
Int[u_.*(a_.+b_.*ArcTan[c_.*x_])^p_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcTan[c*x])^p,x] /;
FreeQ[{a,b,c,p},x] && (EqQ[u,1] ||
  MatchQ[u,(d_.+e_.*x)^q_./; FreeQ[{d,e,q},x]] ||
  MatchQ[u,(f_.*x)^m_.*(d_.+e_.*x)^q_./; FreeQ[{d,e,f,m,q},x]] ||
  MatchQ[u,(d_.+e_.*x^2)^q_./; FreeQ[{d,e,q},x]] ||
  MatchQ[u,(f_.*x)^m_.*(d_.+e_.*x^2)^q_./; FreeQ[{d,e,f,m,q},x]])
```

```
Int[u_.*(a_.+b_.*ArcCot[c_.*x_])^p_.,x_Symbol] :=
  Unintegrable[u*(a+b*ArcCot[c*x])^p,x] /;
FreeQ[{a,b,c,p},x] && (EqQ[u,1] ||
  MatchQ[u,(d_.+e_.*x)^q_./; FreeQ[{d,e,q},x]] ||
  MatchQ[u,(f_.*x)^m_.*(d_.+e_.*x)^q_./; FreeQ[{d,e,f,m,q},x]] ||
  MatchQ[u,(d_.+e_.*x^2)^q_./; FreeQ[{d,e,q},x]] ||
  MatchQ[u,(f_.*x)^m_.*(d_.+e_.*x^2)^q_./; FreeQ[{d,e,f,m,q},x]])
```